
Installation
Note: Phoenix v1.5 comes with built-in support for LiveView apps. Just create
your application with mix phx.new my_app --live. If you are using earlier Phoenix
versions or your app already exists, keep on reading.
The instructions below will serve if you are installing the latest stable version
from Hex. To start using LiveView, add one of the following dependencies to your mix.exs
and run mix deps.get.
If installing from Hex, use the latest version from there:
def deps do
  [
    {:phoenix_live_view, "~> 0.14.0"},
    {:floki, ">= 0.0.0", only: :test}
  ]
end
If you want the latest features, install from GitHub:
def deps do
  [
    {:phoenix_live_view, github: "phoenixframework/phoenix_live_view"},
    {:floki, ">= 0.0.0", only: :test}
  ]
Once installed, update your endpoint's configuration to include a signing salt.
You can generate a signing salt by running mix phx.gen.secret 32:
# config/config.exs

config :my_app, MyAppWeb.Endpoint,
   live_view: [signing_salt: "SECRET_SALT"]
Next, add the following imports to your web file in lib/my_app_web.ex:
# lib/my_app_web.ex

def controller do
  quote do
    ...
    import Phoenix.LiveView.Controller
  end
end

def view do
  quote do
    ...
    import Phoenix.LiveView.Helpers
  end
end

def router do
  quote do
    ...
    import Phoenix.LiveView.Router
  end
end
Then add the Phoenix.LiveView.Router.fetch_live_flash/2 plug to your browser pipeline, in place of :fetch_flash:
# lib/my_app_web/router.ex

pipeline :browser do
  ...
  plug :fetch_session
- plug :fetch_flash
+ plug :fetch_live_flash
end
Next, expose a new socket for LiveView updates in your app's endpoint module.
# lib/my_app_web/endpoint.ex

defmodule MyAppWeb.Endpoint do
  use Phoenix.Endpoint

  # ...

  socket "/live", Phoenix.LiveView.Socket,
    websocket: [connect_info: [session: @session_options]]

  # ...
end
Where @session_options are the options given to plug Plug.Session by using a module attribute. If you don't have a @session_options in your endpoint yet, here is how to create one:
	Find plug Plug.Session in your endpoint.ex

  plug Plug.Session
    store: :cookie,
    key: "_my_app_key",
    signing_salt: "somesigningsalt"
	Move the options to a module attribute at the top of your file:

  @session_options [
    store: :cookie,
    key: "_my_app_key",
    signing_salt: "somesigningsalt"
  ]
	Change the plug Plug.Session to use that attribute:

  plug Plug.Session, @session_options
Add LiveView NPM dependencies to your assets/package.json. For a regular project, do:
{
  "dependencies": {
    "phoenix": "file:../deps/phoenix",
    "phoenix_html": "file:../deps/phoenix_html",
    "phoenix_live_view": "file:../deps/phoenix_live_view"
  }
}
However, if you're adding phoenix_live_view to an umbrella project, the dependency paths should be modified appropriately:
{
  "dependencies": {
    "phoenix": "file:../../../deps/phoenix",
    "phoenix_html": "file:../../../deps/phoenix_html",
    "phoenix_live_view": "file:../../../deps/phoenix_live_view"
  }
}
Then install the new NPM dependency:
npm install --prefix assets
If you had previously installed phoenix_live_view and want to get the
latest javascript, then force an install with:
npm install --force phoenix_live_view --prefix assets
Finally, ensure you have placed a CSRF meta tag inside the <head> tag in your layout (lib/my_app_web/templates/layout/root.html.leex), before app.js is included like so:
<%= csrf_meta_tag() %>
<script type="text/javascript" src="<%= Routes.static_path(@conn, "/js/app.js") %>"></script>
and enable connecting to a LiveView socket in your app.js file.
// assets/js/app.js
import {Socket} from "phoenix"
import LiveSocket from "phoenix_live_view"

let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})

// Connect if there are any LiveViews on the page
liveSocket.connect()

// Expose liveSocket on window for web console debug logs and latency simulation:
// >> liveSocket.enableDebug()
// >> liveSocket.enableLatencySim(1000)
// The latency simulator is enabled for the duration of the browser session.
// Call disableLatencySim() to disable:
// >> liveSocket.disableLatencySim()
window.liveSocket = liveSocket
Layouts
LiveView does not use the default app layout. Instead, you typically call put_root_layout in your router to specify a layout that is used by both "regular" views and live views. In your router, do:
pipeline :browser do
  ...
  plug :put_root_layout, {MyAppWeb.LayoutView, :root}
  ...
end
The layout given to put_root_layout must use <%= @inner_content %> instead of <%= render(@view_module, @view_template, assigns) %>. It is typically very barebones, with mostly
<head> and <body> tags. For example:
<!DOCTYPE html>
<html lang="en">
  <head>
    <%= csrf_meta_tag() %>
    <%= live_title_tag assigns[:page_title] || "MyApp" %>
    <link rel="stylesheet" href="<%%= Routes.static_path(@conn, "/css/app.css") %>"/>
    <script defer type="text/javascript" src="<%%= Routes.static_path(@conn, "/js/app.js") %>"></script>
  </head>
  <body>
    <%%= @inner_content %>
  </body>
</html>
Once you have specified a root layout, "app.html.eex" will be rendered within your root layout for all non-LiveViews. You may also optionally define a "live.html.leex" layout to be used across all LiveViews, as we will describe in the next section.
phx.gen.live support
While the above instructions are enough to install LiveView in a Phoenix app, if you want to use the phx.gen.live generators that come as part of Phoenix v1.5, you need to do one more change, as those generators assume your application was created with mix phx.new --live.
The change is to define the live_view and live_component functions in your my_app_web.ex file, while refactoring the view function. At the end, they will look like this:
  def view do
    quote do
      use Phoenix.View,
        root: "lib/<%= lib_web_name %>/templates",
        namespace: <%= web_namespace %>

      # Import convenience functions from controllers
      import Phoenix.Controller,
        only: [get_flash: 1, get_flash: 2, view_module: 1, view_template: 1]

      # Include shared imports and aliases for views
      unquote(view_helpers())
    end
  end

  def live_view do
    quote do
      use Phoenix.LiveView,
        layout: {<%= web_namespace %>.LayoutView, "live.html"}

      unquote(view_helpers())
    end
  end

  def live_component do
    quote do
      use Phoenix.LiveComponent

      unquote(view_helpers())
    end
  end

  defp view_helpers do
    quote do
      # Use all HTML functionality (forms, tags, etc)
      use Phoenix.HTML

      # Import LiveView helpers (live_render, live_component, live_patch, etc)
      import Phoenix.LiveView.Helpers

      # Import basic rendering functionality (render, render_layout, etc)
      import Phoenix.View

      import MyAppWeb.ErrorHelpers
      import MyAppWeb.Gettext
      alias MyAppWeb.Router.Helpers, as: Routes
    end
  end
Note that LiveViews are automatically configured to use a "live.html.leex" layout in this line:
use Phoenix.LiveView,
  layout: {<%= web_namespace %>.LayoutView, "live.html"}
"root.html.leex" is shared by regular and live views, "app.html.eex" is rendered inside the root layout for regular views, and "live.html.leex" is rendered inside the root layout for LiveViews. "live.html.leex" typically starts out as a copy of "app.html.eex", but using the @socket assign instead of @conn. Check the Live Layouts guide for more information.
Progress animation
If you want to show a progress bar as users perform live actions, we recommend using nprogress.
First add nprogress as a dependency in your assets/package.json:
"nprogress": "^0.2.0"
Then in your assets/css/app.css file, import its style:
@import "../node_modules/nprogress/nprogress.css";
Finally customize LiveView to use it in your assets/js/app.js, right before the liveSocket.connect() call:
import NProgress from "nprogress"

// Show progress bar on live navigation and form submits
window.addEventListener("phx:page-loading-start", info => NProgress.start())
window.addEventListener("phx:page-loading-stop", info => NProgress.done())
  
Assigns and LiveEEx templates
All of the data in a LiveView is stored in the socket as assigns.
The Phoenix.LiveView.assign/2 and Phoenix.LiveView.assign/3
functions help store those values. Those values can be accessed
in the LiveView as socket.assigns.name but they are accessed
inside LiveView templates as @name.
Phoenix.LiveView's built-in templates are identified by the .leex
extension (Live EEx) or ~L sigil. They are similar to regular .eex
templates except they are designed to minimize the amount of data sent
over the wire by splitting static and dynamic parts and tracking changes.
When you first render a .leex template, it will send all of the
static and dynamic parts of the template to the client. After that,
any change you do on the server will now send only the dynamic parts,
and only if those parts have changed.
The tracking of changes is done via assigns. Imagine this template:
<h1><%= expand_title(@title) %></h1>
If the @title assign changes, then LiveView will execute
expand_title(@title) and send the new content. If @title is
the same, nothing is executed and nothing is sent.
Change tracking also works when accessing map/struct fields.
Take this template:
<div id="user_<%= @user.id %>">
  <%= @user.name %>
</div>
If the @user.name changes but @user.id doesn't, then LiveView
will re-render only @user.name and it will not execute or resend @user.id
at all.
The change tracking also works when rendering other templates as
long as they are also .leex templates:
<%= render "child_template.html", assigns %>
The assign tracking feature also implies that you MUST avoid performing
direct operations in the template. For example, if you perform a database
query in your template:
<%= for user <- Repo.all(User) do %>
  <%= user.name %>
<% end %>
Then Phoenix will never re-render the section above, even if the number of
users in the database changes. Instead, you need to store the users as
assigns in your LiveView before it renders the template:
assign(socket, :users, Repo.all(User))
Generally speaking, data loading should never happen inside the template,
regardless if you are using LiveView or not. The difference is that LiveView
enforces this best practice.
LiveEEx pitfalls
There are two common pitfalls to keep in mind when using the ~L sigil
or .leex templates.
When it comes to do/end blocks, change tracking is supported only on blocks
given to Elixir's basic constructs, such as if, case, for, and friends.
If the do/end block is given to a library function or user function, such as
content_tag, change tracking won't work. For example, imagine the following
template that renders a div:
<%= content_tag :div, id: "user_#{@id}" do %>
  <%= @name %>
  <%= @description %>
<% end %>
LiveView knows nothing about content_tag, which means the whole div will
be sent whenever any of the assigns change. This can be easily fixed by
writing the HTML directly:
<div id="user_<%= @id %>">
  <%= @name %>
  <%= @description %>
</div>
Another pitfall of .leex templates is related to variables. Due to the scope
of variables, LiveView has to disable change tracking whenever variables are
used in the template, with the exception of variables introduced by Elixir
basic case, for, and other block constructs. Therefore, you must avoid
code like this in your LiveEEx:
<% some_var = @x + @y %>
<%= some_var %>
Instead, use a function:
<%= sum(@x, @y) %>
Similarly, do not define variables at the top of your render function:
def render(assigns) do
  sum = assigns.x + assigns.y

  ~L"""
  <%= sum %>
  """
end
Instead explicitly precompute the assign in your LiveView, outside of render:
assign(socket, sum: socket.assigns.x + socket.assigns.y)
Generally speaking, avoid accessing variables inside LiveViews. This also applies
to the assigns variable, except when rendering another .leex template. In such
cases, it is ok to pass the whole assigns, as LiveView will continue to perform
change tracking in the called template:
<%= render "sidebar.html", assigns %>
Similarly, variables introduced by Elixir's block constructs are fine. For example,
accessing the post variable defined by the comprehension below works as expected:
<%= for post <- @posts do %>
  ...
<% end %>
As are the variables matched defined in a case or cond:
<%= cond do %>
  <% is_nil(@post) -> %>
    ...
  <% @post -> %>
    ...
<% end %>
To sum up:
	Avoid passing block expressions to library and custom functions

	Never do anything on def render(assigns) besides rendering a template
or invoking the ~L sigil

	Avoid defining local variables, except within for, case, and friends


  
Error and exception handling
As with any other Elixir code, exceptions may happen during the LiveView
life-cycle. In this section we will describe how LiveView reacts to errors
at different stages.
Expected scenarios
In this section, we will talk about error cases that you expect to happen
within your application. For example, a user filling in a form with invalid
data is expected. In a LiveView, we typically handle those cases by storing
a change in the LiveView state, which causes the LiveView to be re-rendered
with the error message.
We may also use flash messages for this. For example, imagine you have a
page to manage all "Team members" in an organization. However, if there is
only one member left in the organization, they should not be allowed to
leave. You may want to handle this by using flash messages:
if MyApp.Org.leave(socket.assigns.current_org, member) do
  {:noreply, socket}
else
  {:noreply, put_flash(socket, :error, "last member cannot leave organization")}
end
However, one may argue that, if the last member of an organization cannot
leave it, it may be better to not even show the "Leave" button in the UI
when the organization has only one member.
Given the button does not appear in the UI, triggering the "leave" when
the organization has now only one member is an unexpected scenario. This
means we can probably rewrite the code above to:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
If leave returns false by any chance, it will just raise. Or you can
even provide a leave! function that raises a specific exception:
MyApp.Org.leave!(socket.assigns.current_org, member)
{:noreply, socket}
However, what will happen with a LiveView in case of exceptions?
Let's talk about unexpected scenarios.
Unexpected scenarios
Elixir developers tend to write assertive code. This means that, if we
expect leave to always return true, we can explicitly match on its
result, as we did above:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
If leave fails and returns false, an exception is raised. It is common
for Elixir developers to use exceptions for unexpected scenarios in their
Phoenix applications.
For example, if you are building an application where a user may belong to
one or more organizations, when accessing the organization page, you may want to
check that the user has access to it like this:
organizations_query = Ecto.assoc(socket.assigns.current_user, :organizations)
Repo.get!(organizations_query, params["org_id"])
The code above builds a query that returns all organizations that belongs to
the current user and then validates that the given "org_id" belongs to the
user. If there is no such "org_id" or if the user has no access to it, an
Ecto.NotFoundError exception is raised.
During a regular controller request, this exception will be converted to a
404 exception and rendered as a custom error page, as
detailed here.
To understand how a LiveView reacts to exceptions, we need to consider two
scenarios: exceptions on mount and during any event.
Exceptions on mount
Given the code on mount runs both on the initial disconnected render and the
connected render, an exception on mount will trigger the following events:
Exceptions during disconnected render:
	An exception on mount is caught and converted to an exception page
by Phoenix error views - pretty much like the way it works with controllers

Exceptions during connected render:
	An exception on mount will crash the LiveView process - which will be logged
	Once the client has noticed the crash during mount, it will fully reload the page
	Reloading the page will start a disconnected render, that will cause mount
to be invoked again and most likely raise the same exception. Except this time
it will be caught and converted to an exception page by Phoenix error views

In other words, LiveView will reload the page in case of errors, making it
fail as if LiveView was not involved in the rendering in the first place.
Exceptions on events (handle_info, handle_event, etc)
If the error happens during an event, the LiveView process will crash. The client
will notice the error and remount the LiveView - without reloading the page. This
is enough to update the page and show the user the latest information.
For example, let's say two users try to leave the organization at the same time.
In this case, both of them see the "Leave" button, but our leave function call
will succeed only for one of them:
true = MyApp.Org.leave(socket.assigns.current_org, member)
{:noreply, socket}
When the exception raises, the client will remount the LiveView. Once you remount,
your code will now notice that there is only one user in the organization and
therefore no longer show the "Leave" button. In other words, by remounting,
we often update the state of the page, allowing exceptions to be automatically
handled.
Note that the choice between conditionally checking on the result of the leave
function with an if, or simply asserting it returns true, is completely
up to you. If the likelihood of everyone leaving the organization at the same
time is low, then you may as well treat it as an unexpected scenario. Although
other developers will be more comfortable by explicitly handling those cases.
In both scenarios, LiveView has you covered.
  
Live layouts
NOTE: Make sure you've read the Assigns and LiveEEx templates
guide before moving forward.
When working with LiveViews, there are usually three layouts to be
considered:
	the root layout - this is a layout used by both LiveView and
regular views. This layout typically contains the <html>
definition alongside the head and body tags. Any content defined
in the root layout will remain the same, even as you live navigate
across LiveViews. All LiveViews defined at the router must have
a root layout. The root layout is typically declared on the
router with put_root_layout and defined as "root.html.eex"
in your MyAppWeb.LayoutView. It may also be given via the
the :layout option to the router's live macro.

	the app layout - this is the default application layout which
is not included or used by LiveViews. It defaults to "app.html.eex"
in your MyAppWeb.LayoutView.

	the live layout - this is the layout which wraps a LiveView and
is rendered as part of the LiveView life-cycle. It must be opt-in
by passing the :layout option on use Phoenix.LiveView. It is
typically set to "live.html.leex"in your MyAppWeb.LayoutView.


Overall, those layouts are found in templates/layout with the
following names:
* root.html.eex
* app.html.eex
* live.html.leex
All layouts must call <%= @inner_content %> to inject the
content rendered by the layout.
The "root" layout is shared by both "app" and "live" layouts.
It is rendered only on the initial request and therefore it
has access to the @conn assign. The root layout must be defined
in your router:
plug :put_root_layout, {MyAppWeb.LayoutView, :root}
Alternatively, the root layout can be passed individually to the
live macro of your live routes:
live "/dashboard", MyAppWeb.Dashboard, layout: {MyAppWeb.LayoutView, :root}
The "app" and "live" layouts are often small and similar to each
other, but the "app" layout uses the @conn and is used as part
of the regular request life-cycle. The "live" layout is part of
the LiveView and therefore has direct access to the @socket.
For example, you can define a new live.html.leex layout with
dynamic content. You must use @inner_content where the output
of the actual template will be placed at:
<p><%= live_flash(@flash, :notice) %></p>
<p><%= live_flash(@flash, :error) %></p>
<%= @inner_content %>
To use the live layout, update your LiveView to pass the :layout
option to use Phoenix.LiveView:
use Phoenix.LiveView, layout: {MyAppWeb.LayoutView, "live.html"}
If you are using Phoenix v1.5, the layout is automatically set
when generating apps with the mix phx.new --live flag.
The :layout option on use does not apply to LiveViews rendered
within other LiveViews. If you want to render child live views or
opt-in to a layout, use :layout as an option in mount:
  def mount(_params, _session, socket) do
    socket = assign(socket, new_message_count: 0)
    {:ok, socket, layout: {MyAppWeb.LayoutView, "live.html"}}
  end
Note: The live layout is always wrapped by the LiveView's :container tag.
Updating the HTML document title
Because the root layout from the Plug pipeline is rendered outside of
LiveView, the contents cannot be dynamically changed. The one exception
is the <title> of the HTML document. Phoenix LiveView special cases
the @page_title assign to allow dynamically updating the title of the
page, which is useful when using live navigation, or annotating the browser
tab with a notification. For example, to update the user's notification
count in the browser's title bar, first set the page_title assign on
mount:
  def mount(_params, _session, socket) do
    socket = assign(socket, page_title: "Latest Posts")
    {:ok, socket}
  end
Then access @page_title in the root layout:
<title><%= @page_title %></title>
You can also use Phoenix.LiveView.Helpers.live_title_tag/2 to support
adding automatic prefix and suffix to the page title when rendered and
on subsequent updates:
<%= live_title_tag @page_title, prefix: "MyApp – " %>
Although the root layout is not updated by LiveView, by simply assigning
to page_title, LiveView knows you want the title to be updated:
def handle_info({:new_messages, count}, socket) do
  {:noreply, assign(socket, page_title: "Latest Posts (#{count} new)")}
end
Note: If you find yourself needing to dynamically patch other parts of the
base layout, such as injecting new scripts or styles into the <head> during
live navigation, then a regular, non-live, page navigation should be used
instead. Assigning the @page_title updates the document.title directly,
and therefore cannot be used to update any other part of the base layout.
  
Live navigation
LiveView provides functionality to allow page navigation using the
browser's pushState API.
With live navigation, the page is updated without a full page reload.
You can trigger live navigation in two ways:
	From the client - this is done by replacing Phoenix.HTML.Link.link/2
by Phoenix.LiveView.Helpers.live_patch/2 or
Phoenix.LiveView.Helpers.live_redirect/2

	From the server - this is done by replacing Phoenix.Controller.redirect/2 calls
by Phoenix.LiveView.push_patch/2 or Phoenix.LiveView.push_redirect/2.


For example, in a template you may write:
<%= live_patch "next", to: Routes.live_path(@socket, MyLive, @page + 1) %>
or in a LiveView:
{:noreply, push_patch(socket, to: Routes.live_path(socket, MyLive, page + 1))}
The "patch" operations must be used when you want to navigate to the
current LiveView, simply updating the URL and the current parameters,
without mounting a new LiveView. When patch is used, the c:handle_params/3
callback is invoked and the minimal set of changes are sent to the client.
See the next section for more information.
The "redirect" operations must be used when you want to dismount the
current LiveView and mount a new one. In those cases, an Ajax request
is made to fetch the necessary information about the new LiveView,
which is mounted in place of the current one within the current layout.
While redirecting, a phx-disconnected class is added to the LiveView,
which can be used to indicate to the user a new page is being loaded.
At the end of the day, regardless if you invoke link/2, live_patch/2,
and live_redirect/2 from the client, or redirect/2, push_patch/2,
and push_redirect/2 from the server, the user will end-up on the same
page. The difference between those is mostly the amount of data sent over
the wire:
	link/2 and redirect/2 do full page reloads

	live_redirect/2 and push_redirect/2 mounts a new LiveView while
keeping the current layout

	live_patch/2 and push_patch/2 updates the current LiveView and
sends only the minimal diff while also maintaining the scroll position


An easy rule of thumb is to stick with live_redirect/2 and push_redirect/2
and use the patch helpers only in the cases where you want to minimize the
amount of data sent when navigating within the same LiveView (for example,
if you want to change the sorting of a table while also updating the URL).
handle_params/3
The c:handle_params/3 callback is invoked after c:mount/3 and before
the initial render. It is also invoked every time live_patch/2 or
push_patch/2 are used. It receives the request parameters as first
argument, the url as second, and the socket as third.
For example, imagine you have a UserTable LiveView to show all users in
the system and you define it in the router as:
live "/users", UserTable
Now to add live sorting, you could do:
<%= live_patch "Sort by name", to: Routes.live_path(@socket, UserTable, %{sort_by: "name"}) %>
When clicked, since we are navigating to the current LiveView, c:handle_params/3
will be invoked. Remember you should never trust the received params, so you must
use the callback to validate the user input and change the state accordingly:
def handle_params(params, _uri, socket) do
  socket =
    case params["sort_by"] do
      sort_by when sort_by in ~w(name company) -> assign(socket, sort_by: sort)
      _ -> socket
    end

  {:noreply, load_users(socket)}
end
As with other handle_* callback, changes to the state inside c:handle_params/3
will trigger a server render.
Note the parameters given to c:handle_params/3 are the same as the ones given
to c:mount/3. So how do you decide which callback to use to load data?
Generally speaking, data should always be loaded on c:mount/3, since c:mount/3
is invoked once per LiveView life-cycle. Only the params you expect to be changed
via live_patch/2 or push_patch/2 must be loaded on c:handle_params/3.
For example, imagine you have a blog. The URL for a single post is:
"/blog/posts/:post_id". In the post page, you have comments and they are paginated.
You use live_patch/2 to update the shown comments every time the user paginates,
updating the URL to "/blog/posts/:post_id?page=X". In this example, you will access
"post_id" on c:mount/3 and the page of comments on c:handle_params/3.
Furthermore, it is very important to not access the same parameters on both
c:mount/3 and c:handle_params/3. For example, do NOT do this:
def mount(%{"post_id" => post_id}, session, socket) do
  # do something with post_id
end

def handle_params(%{"post_id" => post_id, "page" => page}, url, socket) do
  # do something with post_id and page
end
If you do that, because c:mount/3 is called once and c:handle_params/3 multiple
times, the "post_id" read on mount can get out of sync with the one in c:handle_params/3.
So once a parameter is read on mount, it should not be read elsewhere. Instead, do this:
def mount(%{"post_id" => post_id}, session, socket) do
  # do something with post_id
end

def handle_params(%{"sort_by" => sort_by}, url, socket) do
  post_id = socket.assigns.post.id
  # do something with sort_by
end
Replace page address
LiveView also allows the current browser URL to be replaced. This is useful when you
want certain events to change the URL but without polluting the browser's history.
This can be done by passing the replace: true option to any of the navigation helpers.
Multiple LiveViews in the same page
LiveView allows you to have multiple LiveViews in the same page by calling
Phoenix.LiveView.Helpers.live_render/3 in your templates. However, only
the LiveViews defined directly in your router can use the "Live Navigation"
functionality described here. This is important because LiveViews work
closely with your router, guaranteeing you can only navigate to known
routes.
  
Security considerations of the LiveView model
As we have seen, LiveView begins its life-cycle as a regular HTTP request.
Then a stateful connection is established. Both the HTTP request and
the stateful connection receives the client data via parameters and session.
This means that any session validation must happen both in the HTTP request
and the stateful connection.
Mounting considerations
For example, if your HTTP request perform user authentication and confirmation
on every request via Plugs, such as this:
plug :ensure_user_authenticated
plug :ensure_user_confirmed
Then the c:mount/3 callback of your LiveView should execute those same
verifications:
def mount(params, %{"user_id" => user_id} = _session, socket) do
  socket = assign(socket, current_user: Accounts.get_user!(user_id))

  socket =
    if socket.assigns.current_user.confirmed_at do
      socket
    else
      redirect(socket, to: "/login")
    end

  {:ok, socket}
end
Given almost all c:mount/3 actions in your application will have to
perform these exact steps, we recommend creating a function called
assign_defaults/2 or similar, putting it in a new module like
MyAppWeb.LiveHelpers, and modifying lib/my_app_web.ex so all
LiveViews automatically import it:
def live_view do
  quote do
    # ...other stuff...
    import MyAppWeb.LiveHelpers
  end
end
Then make sure to call it in every LiveView's c:mount/3:
def mount(params, session, socket) do
  {:ok, assign_defaults(session, socket)}
end
Where MyAppWeb.LiveHelpers can be something like:
defmodule MyAppWeb.LiveHelpers do
  import Phoenix.LiveView

  def assign_defaults(%{"user_id" => user_id}, socket) do
    socket = assign(socket, current_user: Accounts.get_user!(user_id))

    if socket.assigns.current_user.confirmed_at do
      socket
    else
      redirect(socket, to: "/login")
    end
  end
end
One possible concern in this approach is that in regular HTTP requests the
current user will be fetched twice: one in the HTTP request and another on
mount. You can address this by using the assign_new function, that will
reuse any of the connection assigns from the HTTP request:
def assign_defaults(%{"user_id" => user_id}, socket) do
  socket = assign_new(socket, :current_user, fn -> Accounts.get_user!(user_id) end)

  if socket.assigns.current_user.confirmed_at do
    socket
  else
    redirect(socket, to: "/login")
  end
end
Events considerations
It is also important to keep in mind that LiveView are stateful. Therefore,
if you load any data on c:mount/3 and the data itself changes, the data
won't be automatically propagated to the LiveView, unless you broadcast
those events with Phoenix.PubSub.
Generally speaking, the simplest and safest approach is to perform authorization
whenever there is an action. For example, imagine that you have a LiveView
for a "Blog", and only editors can edit posts. Therefore, it is best to validate
the user is an editor on mount and on every event:
def mount(%{"post_id" => post_id}, session, socket) do
  socket = assign_defaults(session, socket)
  post = Blog.get_post_for_user!(socket.assigns.current_user, post_id)
  {:ok, assign(socket, post: post)}
end

def handle_event("update_post", params, socket) do
  updated_post = Blog.update_post(socket.assigns.current_user, socket.assigns.post, params)
  {:noreply, assign(socket, post: updated_post)}
end
In the example above, the Blog context receives the user on both get and
update operations, and always validates accordingly that the user has access,
raising an error otherwise.
Disconnecting all instances of a given live user
Another security consideration is how to disconnect all instances of a given
live user. For example, imagine the user logs outs, its account is terminated,
or any other reason.
Luckily, it is possible to identify all LiveView sockets by setting a "live_socket_id"
in the session. For example, when signing in a user, you could do:
conn
|> put_session(:current_user_id, user.id)
|> put_session(:live_socket_id, "users_socket:#{user.id}")
Now all LiveView sockets will be identified and listening to the given
live_socket_id. You can disconnect all live users identified by said
ID by broadcasting on the topic:
MyAppWeb.Endpoint.broadcast("users_socket:#{user.id}", "disconnect", %{})
Once a LiveView is disconnected, the client will attempt to reestablish
the connection, re-executing the c:mount/3 callback. In this case,
if the user is no longer logged in or it no longer has access to its
current resource, c:mount/3 will fail and the user will be redirected
to the proper page.
This is the same mechanism provided by Phoenix.Channels. Therefore, if
your application uses both channels and LiveViews, you can use the same
technique to disconnect any stateful connection.
  
Telemetry
LiveView currently exposes the following telemetry events:
	[:phoenix, :live_view, :mount, :start] - Dispatched by a Phoenix.LiveView immediately before c:mount/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params | :not_mounted_at_router,
  session: map
}



	[:phoenix, :live_view, :mount, :stop] - Dispatched by a Phoenix.LiveView when the c:mount/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params | :not_mounted_at_router,
  session: map
}



	[:phoenix, :live_view, :mount, :exception] - Dispatched by a Phoenix.LiveView when an exception is raised in the c:mount/3 callback.
	Measurement: %{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  params: unsigned_params | :not_mounted_at_router,
  session: map
}



	[:phoenix, :live_view, :handle_params, :start] - Dispatched by a Phoenix.LiveView immediately before c:handle_params/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_params, :stop] - Dispatched by a Phoenix.LiveView when the c:handle_params/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_params, :exception] - Dispatched by a Phoenix.LiveView when the when an exception is raised in the c:handle_params/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  params: unsigned_params,
  uri: String.t()
}



	[:phoenix, :live_view, :handle_event, :start] - Dispatched by a Phoenix.LiveView immediately before c:handle_event/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_view, :handle_event, :stop] - Dispatched by a Phoenix.LiveView when the c:handle_event/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_view, :handle_event, :exception] - Dispatched by a Phoenix.LiveView when an exception is raised in the c:handle_event/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :handle_event, :start] - Dispatched by a Phoenix.LiveComponent immediately before c:handle_event/3 is invoked.
	Measurement:
%{system_time: System.monotonic_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :handle_event, :stop] - Dispatched by a Phoenix.LiveComponent when the c:handle_event/3 callback completes successfully.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  component: atom,
  event: String.t(),
  params: unsigned_params
}



	[:phoenix, :live_component, :handle_event, :exception] - Dispatched by a Phoenix.LiveComponent when an exception is raised in the c:handle_event/3 callback.
	Measurement:
%{duration: native_time}

	Metadata:
%{
  socket: Phoenix.LiveView.Socket.t,
  kind: atom,
  reason: term,
  component: atom,
  event: String.t(),
  params: unsigned_params
}




  
Using Gettext for internationalization
For internationalization with gettext,
the locale used within your Plug pipeline can be stored in the Plug session and
restored within your LiveView mount. For example, after user signs in or preference
changes, you can write the locale to the session:
def put_user_session(conn, current_user) do
  locale = get_locale_for_user(current_user)
  Gettext.put_locale(MyApp.Gettext, locale)

  conn
  |> put_session(:user_id, current_user.id)
  |> put_session(:locale, locale)
end
Then in your LiveView mount/3, you can restore the locale:
def mount(_params, %{"locale" => locale}, socket) do
  Gettext.put_locale(MyApp.Gettext, locale)
  {:ok socket}
end
  
Bindings
Phoenix supports DOM element bindings for client-server interaction. For
example, to react to a click on a button, you would render the element:
<button phx-click="inc_temperature">+</button>
Then on the server, all LiveView bindings are handled with the handle_event
callback, for example:
def handle_event("inc_temperature", _value, socket) do
  {:ok, new_temp} = Thermostat.inc_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end
	Binding	Attributes
	Params	phx-value-*
	Click Events	phx-capture-click, phx-click
	Focus/Blur Events	phx-window-focus, phx-window-blur, phx-focus, phx-blur
	Key Events	phx-window-keyup, phx-window-keydown, phx-keyup, phx-keydown
	Form Events	phx-auto-recover, phx-trigger-action, phx-disable-with, phx-feedback-for, phx-submit, phx-change
	Rate Limiting	phx-throttle, phx-debounce
	DOM Patching	phx-update
	JS Interop	phx-hook

Click Events
The phx-click binding is used to send click events to the server.
When any client event, such as a phx-click click is pushed, the value
sent to the server will be chosen with the following priority:
	Any number of optional phx-value- prefixed attributes, such as:
<div phx-click="inc" phx-value-myvar1="val1" phx-value-myvar2="val2">
will send the following map of params to the server:
def handle_event("inc", %{"myvar1" => "val1", "myvar2" => "val2"}, socket) do
If the phx-value- prefix is used, the server payload will also contain a "value"
if the element's value attribute exists.

	When receiving a map on the server, the payload will also include user defined metadata
of the client event, or an empty map if none is set. For example, the following LiveSocket
client option would send the coordinates and altKey information for all clicks:
let liveSocket = new LiveSocket("/live", Socket, {
  params: {_csrf_token: csrfToken},
  metadata: {
    click: (e, el) => {
      return {
        altKey: e.altKey,
        clientX: e.clientX,
        clientY: e.clientY
      }
    }
  }
})


The phx-capture-click event is just like phx-click, but instead of the click event
being dispatched to the closest phx-click element as it bubbles up through the DOM, the event
is dispatched as it propagates from the top of the DOM tree down to the target element. This is
useful when wanting to bind click events without receiving bubbled events from child UI elements.
Since capturing happens before bubbling, this can also be important for preparing or preventing
behaviour that will be applied during the bubbling phase.
Focus and Blur Events
Focus and blur events may be bound to DOM elements that emit
such events, using the phx-blur, and phx-focus bindings, for example:
<input name="email" phx-focus="myfocus" phx-blur="myblur"/>
To detect when the page itself has received focus or blur,
phx-window-focus and phx-window-blur may be specified. These window
level events may also be necessary if the element in consideration
(most often a div with no tabindex) cannot receive focus. Like other
bindings, phx-value-* can be provided on the bound element, and those
values will be sent as part of the payload. For example:
<div class="container"
    phx-window-focus="page-active"
    phx-window-blur="page-inactive"
    phx-value-page="123">
  ...
</div>
The following window-level bindings are supported:
	phx-window-focus
	phx-window-blur
	phx-window-keydown
	phx-window-keyup

Key Events
The onkeydown, and onkeyup events are supported via the phx-keydown,
and phx-keyup bindings. Each binding supports a phx-key attribute, which triggers
the event for the specific key press. If no phx-key is provided, the event is triggered
for any key press. When pushed, the value sent to the server will contain the "key"
that was pressed, plus any user-defined metadata. For example, pressing the
Escape key looks like this:
%{"key" => "Escape"}
To capture additional user-defined metadata, the metadata option for keydown events
may be provided to the LiveSocket constructor. For example:
let liveSocket = new LiveSocket("/live", Socket, {
  params: {_csrf_token: csrfToken},
  metadata: {
    keydown: (e, el) => {
      return {
        key: e.key,
        metaKey: e.metaKey,
        repeat: e.repeat
      }
    }
  }
})
To determine which key has been pressed you should use key value. The
available options can be found on
MDN
or via the Key Event Viewer.
By default, the bound element will be the event listener, but a
window-level binding may be provided via phx-window-keydown or phx-window-keyup,
for example:
def render(assigns) do
  ~L"""
  <div id="thermostat" phx-window-keyup="update_temp">
    Current temperature: <%= @temperature %>
  </div>
  """
end

def handle_event("update_temp", %{"key" => "ArrowUp"}, socket) do
  {:ok, new_temp} = Thermostat.inc_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end

def handle_event("update_temp", %{"key" => "ArrowDown"}, socket) do
  {:ok, new_temp} = Thermostat.dec_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end

def handle_event("update_temp", _key, socket) do
  {:noreply, socket}
end
Rate limiting events with Debounce and Throttle
All events can be rate-limited on the client by using the
phx-debounce and phx-throttle bindings, with the following behavior:
	phx-debounce - Accepts either a string integer timeout value, or "blur".
When an int is provided, delays emitting the event by provided milliseconds.
When "blur" is provided, delays emitting an input's change event until the
field is blurred by the user. Debounce is typically emitted for inputs.

	phx-throttle - Accepts an integer timeout value to throttle the event in milliseconds.
Unlike debounce, throttle will immediately emit the event, then rate limit the
event at one event per provided timeout. Throttle is typically used to rate limit
clicks, mouse and keyboard actions.


For example, to avoid validating an email until the field is blurred, while validating
the username at most every 2 seconds after a user changes the field:
<form phx-change="validate" phx-submit="save">
  <input type="text" name="user[email]" phx-debounce="blur"/>
  <input type="text" name="user[username]" phx-debounce="2000"/>
</form>
And to rate limit a volume up click to once every second:
<button phx-click="volume_up" phx-throttle="1000">+</button>
Likewise, you may throttle held-down keydown:
<div phx-window-keydown="keydown" phx-throttle="500">
  ...
</div>
Unless held-down keys are required, a better approach is generally to use
phx-keyup bindings which only trigger on key up, thereby being self-limiting.
However, phx-keydown is useful for games and other use cases where a constant
press on a key is desired. In such cases, throttle should always be used.
Debounce and Throttle special behavior
The following specialized behavior is performed for forms and keydown bindings:
	When a phx-submit, or a phx-change for a different input is triggered,
any current debounce or throttle timers are reset for existing inputs.

	A phx-keydown binding is only throttled for key repeats. Unique keypresses
back-to-back will dispatch the pressed key events.


LiveView Specific Events
The lv: event prefix supports LiveView specific features that are handled
by LiveView without calling the user's handle_event/3 callbacks. Today,
the following events are supported:
	lv:clear-flash – clears the flash when sent to the server. If a
phx-value-key is provided, the specific key will be removed from the flash.

For example:
<p class="alert" phx-click="lv:clear-flash" phx-value-key="info">
  <%= live_flash(@flash, :info) %>
</p>
  
Form bindings
Form Events
To handle form changes and submissions, use the phx-change and phx-submit
events. In general, it is preferred to handle input changes at the form level,
where all form fields are passed to the LiveView's callback given any
single input change. For example, to handle real-time form validation and
saving, your template would use both phx_change and phx_submit bindings:
<%= f = form_for @changeset, "#", [phx_change: :validate, phx_submit: :save] %>
  <%= label f, :username %>
  <%= text_input f, :username %>
  <%= error_tag f, :username %>

  <%= label f, :email %>
  <%= text_input f, :email %>
  <%= error_tag f, :email %>

  <%= submit "Save" %>
</form>
Reminder: form_for/3 is a Phoenix.HTML helper. Don't forget to include
use Phoenix.HTML at the top of your LiveView, if using Phoenix.HTML helpers.
Also, if using error_tag/2, don't forget to import MyAppWeb.ErrorHelpers.
Next, your LiveView picks up the events in handle_event callbacks:
def render(assigns) ...

def mount(_params, _session, socket) do
  {:ok, assign(socket, %{changeset: Accounts.change_user(%User{})})}
end

def handle_event("validate", %{"user" => params}, socket) do
  changeset =
    %User{}
    |> Accounts.change_user(params)
    |> Map.put(:action, :insert)

  {:noreply, assign(socket, changeset: changeset)}
end

def handle_event("save", %{"user" => user_params}, socket) do
  case Accounts.create_user(user_params) do
    {:ok, user} ->
      {:noreply,
       socket
       |> put_flash(:info, "user created")
       |> redirect(to: Routes.user_path(MyAppWeb.Endpoint, MyAppWeb.User.ShowView, user))}

    {:error, %Ecto.Changeset{} = changeset} ->
      {:noreply, assign(socket, changeset: changeset)}
  end
end
The validate callback simply updates the changeset based on all form input
values, then assigns the new changeset to the socket. If the changeset
changes, such as generating new errors, c:render/1 is invoked and
the form is re-rendered.
Likewise for phx-submit bindings, the same callback is invoked and
persistence is attempted. On success, a :noreply tuple is returned and the
socket is annotated for redirect with Phoenix.LiveView.redirect/2 to
the new user page, otherwise the socket assigns are updated with the errored
changeset to be re-rendered for the client.
phx-feedback-for
For proper form error tag updates, the error tag must specify which
input it belongs to. This is accomplished with the phx-feedback-for attribute.
Failing to add the phx-feedback-for attribute will result in displaying error
messages for form fields that the user has not changed yet (e.g. required
fields further down on the page).
For example, your MyAppWeb.ErrorHelpers may use this function:
def error_tag(form, field) do
  form.errors
  |> Keyword.get_values(field)
  |> Enum.map(fn error ->
    content_tag(:span, translate_error(error),
      class: "invalid-feedback",
      phx_feedback_for: input_id(form, field)
    )
  end)
end
Now, any DOM container with the phx-feedback-for attribute will receive a
phx-no-feedback class in cases where the form fields has yet to receive
user input/focus. The following css rules are generated in new projects
to hide the errors:
.phx-no-feedback.invalid-feedback, .phx-no-feedback .invalid-feedback {
  display: none;
}
Number inputs
Number inputs are a special case in LiveView forms. On programmatic updates,
some browsers will clear invalid inputs. So LiveView will not send change events
from the client when an input is invalid, instead allowing the browser's native
validation UI to drive user interaction. Once the input becomes valid, change and
submit events will be sent normally.
<input type="number">
This is known to have a plethora of problems including accessibility, large numbers
are converted to exponential notation and scrolling can accidentally increase or
decrease the number.
As of early 2020, the following avoids these pitfalls and will likely serve your
application's needs and users much better. According to https://caniuse.com/#search=inputmode,
the following is supported by 90% of the global mobile market with Firefox yet to implement.
<input type="text" inputmode="numeric" pattern="[0-9]*">
Password inputs
Password inputs are also special cased in Phoenix.HTML. For security reasons,
password field values are not reused when rendering a password input tag. This
requires explicitly setting the :value in your markup, for example:
<%= password_input f, :password, value: input_value(f, :password) %>
<%= password_input f, :password_confirmation, value: input_value(f, :password_confirmation) %>
<%= error_tag f, :password %>
<%= error_tag f, :password_confirmation %>
Submitting the form action over HTTP
The phx-trigger-action attribute can be added to a form to trigger a standard
form submit on DOM patch to the URL specified in the form's standard action
attribute. This is useful to perform pre-final validation of a LiveView form
submit before posting to a controller route for operations that require
Plug session mutation. For example, in your LiveView template you can
annotate the phx-trigger-action with a boolean assign:
<%= f = form_for @changeset, Routes.reset_password_path(@socket, :create),
  phx_submit: :save,
  phx_trigger_action: @trigger_submit %>
Then in your LiveView, you can toggle the assign to trigger the form with the current
fields on next render:
def handle_event("save", params, socket) do
  case validate_change_password(socket.assigns.user, params) do
    {:ok, changeset} ->
      {:noreply, assign(socket, changeset: changeset, trigger_submit: true)}

    {:error, changeset} ->
      {:noreply, assign(socket, changeset: changeset)}
    end
end
Recovery following crashes or disconnects
By default, all forms marked with phx-change will recover input values
automatically after the user has reconnected or the LiveView has remounted
after a crash. This is achieved by the client triggering the same phx-change
to the server as soon as the mount has been completed.
Note: if you want to see form recovery working in development, please
make sure to disable live reloading in development by commenting out the
LiveReload plug in your endpoint.ex file or by setting code_reloader: false
in your config/dev.exs. Otherwise live reloading may cause the current page
to be reloaded whenever you restart the server, which will discard all form
state.
For most use cases, this is all you need and form recovery will happen
without consideration. In some cases, where forms are built step-by-step in a
stateful fashion, it may require extra recovery handling on the server outside
of your existing phx-change callback code. To enable specialized recovery,
provide a phx-auto-recover binding on the form to specify a different event
to trigger for recovery, which will receive the form params as usual. For example,
imagine a LiveView wizard form where the form is stateful and built based on what
step the user is on and by prior selections:
<form phx-change="validate_wizard_step" phx-auto-recover="recover_wizard">
On the server, the "validate_wizard_step" event is only concerned with the
current client form data, but the server maintains the entire state of the wizard.
To recover in this scenario, you can specify a recovery event, such as "recover_wizard"
above, which would wire up to the following server callbacks in your LiveView:
def handle_event("validate_wizard_step", params, socket) do
  # regular validations for current step
  {:noreply, socket}
end

def handle_event("recover_wizard", params, socket) do
  # rebuild state based on client input data up to the current step
  {:noreply, socket}
end
To forgo automatic form recovery, set phx-auto-recover="ignore".
JavaScript client specifics
The JavaScript client is always the source of truth for current input values.
For any given input with focus, LiveView will never overwrite the input's current
value, even if it deviates from the server's rendered updates. This works well
for updates where major side effects are not expected, such as form validation
errors, or additive UX around the user's input values as they fill out a form.
For these use cases, the phx-change input does not concern itself with disabling
input editing while an event to the server is in flight. When a phx-change event
is sent to the server, the input tag and parent form tag receive the
phx-change-loading css class, then the payload is pushed to the server with a
"_target" param in the root payload containing the keyspace of the input name
which triggered the change event.
For example, if the following input triggered a change event:
<input name="user[username]"/>
The server's handle_event/3 would receive a payload:
%{"_target" => ["user", "username"], "user" => %{"username" => "Name"}}
The phx-submit event is used for form submissions where major side effects
typically happen, such as rendering new containers, calling an external
service, or redirecting to a new page.
On submission of a form bound with a phx-submit event:
	The form's inputs are set to readonly
	Any submit button on the form is disabled
	The form receives the "phx-submit-loading" class

On completion of server processing of the phx-submit event:
	The submitted form is reactivated and loses the "phx-submit-loading" class
	The last input with focus is restored (unless another input has received focus)
	Updates are patched to the DOM as usual

To handle latent events, any HTML tag can be annotated with
phx-disable-with, which swaps the element's innerText with the provided
value during event submission. For example, the following code would change
the "Save" button to "Saving...", and restore it to "Save" on acknowledgment:
<button type="submit" phx-disable-with="Saving...">Save</button>
You may also take advantage of LiveView's CSS loading state classes to
swap out your form content while the form is submitting. For example,
with the following rules in your app.css:
.while-submitting { display: none; }
.inputs { display: block; }

.phx-submit-loading {
  .while-submitting { display: block; }
  .inputs { display: none; }
}
You can show and hide content with the following markup:
<form phx-change="update">
  <div class="while-submitting">Please wait while we save our content...</div>
  <div class="inputs">
    <input type="text" name="text" value="<%= @text %>">
  </div>
</form>
Additionally, we strongly recommend including a unique HTML "id" attribute on the form.
When DOM siblings change, elements without an ID will be replaced rather than moved,
which can cause issues such as form fields losing focus.
  
DOM patching and temporary assigns
A container can be marked with phx-update, allowing the DOM patch
operations to avoid updating or removing portions of the LiveView, or to append
or prepend the updates rather than replacing the existing contents. This
is useful for client-side interop with existing libraries that do their
own DOM operations. The following phx-update values are supported:
	replace - the default operation. Replaces the element with the contents
	ignore - ignores updates to the DOM regardless of new content changes
	append - append the new DOM contents instead of replacing
	prepend - prepend the new DOM contents instead of replacing

When using phx-update, a unique DOM ID must always be set in the
container. If using "append" or "prepend", a DOM ID must also be set
for each child. When appending or prepending elements containing an
ID already present in the container, LiveView will replace the existing
element with the new content instead appending or prepending a new
element.
The "ignore" behaviour is frequently used when you need to integrate
with another JS library. The "append" and "prepend" feature is often
used with "Temporary assigns" to work with large amounts of data. Let's
learn more.
Temporary assigns
By default, all LiveView assigns are stateful, which enables change
tracking and stateful interactions. In some cases, it's useful to mark
assigns as temporary, meaning they will be reset to a default value after
each update. This allows otherwise large but infrequently updated values
to be discarded after the client has been patched.
Imagine you want to implement a chat application with LiveView. You
could render each message like this:
<%= for message <- @messages do %>
  <p><span><%= message.username %>:</span> <%= message.text %></p>
<% end %>
Every time there is a new message, you would append it to the @messages
assign and re-render all messages.
As you may suspect, keeping the whole chat conversation in memory
and resending it on every update would be too expensive, even with
LiveView smart change tracking. By using temporary assigns and phx-update,
we don't need to keep any messages in memory, and send messages to be
appended to the UI only when there are new ones.
To do so, the first step is to mark which assigns are temporary and
what values they should be reset to on mount:
def mount(_params, _session, socket) do
  socket = assign(socket, :messages, load_last_20_messages())
  {:ok, socket, temporary_assigns: [messages: []]}
end
On mount we also load the initial number of messages we want to
send. After the initial render, the initial batch of messages will
be reset back to an empty list.
Now, whenever there are one or more new messages, we will assign
only the new messages to @messages:
socket = assign(socket, :messages, new_messages)
In the template, we want to wrap all of the messages in a container
and tag this content with phx-update. Remember, we must add an ID
to the container as well as to each child:
<div id="chat-messages" phx-update="append">
  <%= for message <- @messages do %>
    <p id="<%= message.id %>">
      <span><%= message.username %>:</span> <%= message.text %>
    </p>
  <% end %>
</div>
When the client receives new messages, it now knows to append to the
old content rather than replace it.
  
JavaScript interoperability
As seen earlier, you start by instantiating a single LiveSocket to enable LiveView
client/server interaction, for example:
import {Socket} from "phoenix"
import LiveSocket from "phoenix_live_view"

let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})
liveSocket.connect()
All options are passed directly to the Phoenix.Socket constructor,
except for the following LiveView specific options:
	bindingPrefix - the prefix to use for phoenix bindings. Defaults "phx-"
	params - the connect_params to pass to the view's mount callback. May be
a literal object or closure returning an object. When a closure is provided,
the function receives the view's phx-view name.
	hooks – a reference to a user-defined hooks namespace, containing client
callbacks for server/client interop. See the interop section below for details.

Debugging Client Events
To aid debugging on the client when troubleshooting issues, the enableDebug()
and disableDebug() functions are exposed on the LiveSocket JavaScript instance.
Calling enableDebug() turns on debug logging which includes LiveView life-cycle and
payload events as they come and go from client to server. In practice, you can expose
your instance on window for quick access in the browser's web console, for example:
// app.js
let liveSocket = new LiveSocket(...)
liveSocket.connect()
window.liveSocket = liveSocket

// in the browser's web console
>> liveSocket.enableDebug()
The debug state uses the browser's built-in sessionStorage, so it will remain in effect
for as long as your browser session lasts.
Simulating Latency
Proper handling of latency is critical for good UX. LiveView's CSS loading states allow
the client to provide user feedback while awaiting a server response. In development,
near zero latency on localhost does not allow latency to be easily represented or tested,
so LiveView includes a latency simulator with the JavaScript client to ensure your
application provides a pleasant experience. Like the enableDebug() function above,
the LiveSocket instance includes enableLatencySim(milliseconds) and disableLatencySim()
functions which apply throughout the current browser session. The enableLatencySim function
accepts an integer in milliseconds for the round-trip-time to the server. For example:
// app.js
let liveSocket = new LiveSocket(...)
liveSocket.connect()
window.liveSocket = liveSocket

// in the browser's web console
>> liveSocket.enableLatencySim(1000)
[Log] latency simulator enabled for the duration of this browser session.
      Call disableLatencySim() to disable
Loading state and errors
By default, the following classes are applied to the LiveView's parent
container:
	"phx-connected" - applied when the view has connected to the server
	"phx-disconnected" - applied when the view is not connected to the server
	"phx-error" - applied when an error occurs on the server. Note, this
class will be applied in conjunction with "phx-disconnected" if connection
to the server is lost.

All phx- event bindings apply their own css classes when pushed. For example
the following markup:
<button phx-click="clicked" phx-window-keydown="key">...</button>
On click, would receive the phx-click-loading class, and on keydown would receive
the phx-keydown-loading class. The css loading classes are maintained until an
acknowledgement is received on the client for the pushed event.
In the case of forms, when a phx-change is sent to the server, the input element
which emitted the change receives the phx-change-loading class, along with the
parent form tag. The following events receive css loading classes:
	phx-click - phx-click-loading
	phx-change - phx-change-loading
	phx-submit - phx-submit-loading
	phx-focus - phx-focus-loading
	phx-blur - phx-blur-loading
	phx-window-keydown - phx-keydown-loading
	phx-window-keyup - phx-keyup-loading

For live page navigation via live_redirect and live_patch, as well as form
submits via phx-submit, the JavaScript events "phx:page-loading-start" and
"phx:page-loading-stop" are dispatched on window. Additionally, any phx-
event may dispatch page loading events by annotating the DOM element with
phx-page-loading. This is useful for showing main page loading status, for example:
// app.js
import NProgress from "nprogress"
window.addEventListener("phx:page-loading-start", info => NProgress.start())
window.addEventListener("phx:page-loading-stop", info => NProgress.done())
Within the callback, info.detail will be an object that contains a kind
key, with a value that depends on the triggering event:
	"redirect" - the event was triggered by a redirect
	"patch" - the event was triggered by a patch
	"initial" - the event was triggered by initial page load
	"element" - the event was triggered by a phx- bound element, such as phx-click

For all kinds of page loading events, all but "element" will receive an additional to
key in the info metadata pointing to the href associated with the page load.
In the case of an "element" page loading event, the info will contain a
"target" key containing the DOM element which triggered the page loading
state.
Client hooks
To handle custom client-side JavaScript when an element is added, updated,
or removed by the server, a hook object may be provided with the following
life-cycle callbacks:
	mounted - the element has been added to the DOM and its server
LiveView has finished mounting
	beforeUpdate - the element is about to be updated in the DOM.
Note: any call here must be synchronous as the operation cannot
be deferred or cancelled.
	updated - the element has been updated in the DOM by the server
	beforeDestroy - the element is about to be removed from the DOM.
Note: any call here must be synchronous as the operation cannot
be deferred or cancelled.
	destroyed - the element has been removed from the page, either
by a parent update, or by the parent being removed entirely
	disconnected - the element's parent LiveView has disconnected from the server
	reconnected - the element's parent LiveView has reconnected to the server

The above life-cycle callbacks have in-scope access to the following attributes:
	el - attribute referencing the bound DOM node,
	viewName - attribute matching the DOM node's phx-view value
	pushEvent(event, payload, (reply, ref) => ...) - method to push an event from the client to the LiveView server
	pushEventTo(selectorOrTarget, event, payload, (reply, ref) => ...) - method to push targeted events from the client
to LiveViews and LiveComponents.
	handleEvent(event, (payload) => ...) - method to handle an event pushed from the server

For example, the markup for a controlled input for phone-number formatting could be written
like this:
<input type="text" name="user[phone_number]" id="user-phone-number" phx-hook="PhoneNumber" />
Then a hook callback object could be defined and passed to the socket:
let Hooks = {}
Hooks.PhoneNumber = {
  mounted() {
    this.el.addEventListener("input", e => {
      let match = this.el.value.replace(/\D/g, "").match(/^(\d{3})(\d{3})(\d{4})$/)
      if(match) {
        this.el.value = `${match[1]}-${match[2]}-${match[3]}`
      }
    })
  }
}

let liveSocket = new LiveSocket("/live", Socket, {hooks: Hooks, ...})
...
The hook can push events to the LiveView by using the pushEvent function and receive a
reply from the server via a {:reply, map, socket} return value. The reply payload will be
passed to the optional pushEvent response callback.
Communication with the hook from the server can be done by reading data attributes on the
hook element, or by using push_event on the server and handleEvent on the client.
For example, to implement infinite scrolling, one might do:
<div id="infinite-scroll" phx-hook="InfiniteScroll" data-page="<%= @page %>">
And then in the client:
Hooks.InfiniteScroll = {
  page() { return this.el.dataset.page },
  mounted(){
    this.pending = this.page()
    window.addEventListener("scroll", e => {
      if(this.pending == this.page() && scrollAt() > 90){
        this.pending = this.page() + 1
        this.pushEvent("load-more", {})
      }
    })
  },
  updated(){ this.pending = this.page() }
}
To push out-of-band events to the client, for example to render charting points, one could do:
<div id="chart" phx-hook="Chart">

{:noreply, push_event(socket, "points", %{points: new_points})}
And then on the client:
Hooks.Chart = {
  mounted(){
    this.handleEvent("points", ({points}) => MyChartLib.addPoints(points))
  }
}
Note: events pushed from the server via push_event are global and will be dispatched
to all active hooks on the client who are handling that event.
Note: when using phx-hook, a unique DOM ID must always be set.
For integration with client-side libraries which require a broader access to full
DOM management, the LiveSocket constructor accepts a dom option with an
onBeforeElUpdated callback. The fromEl and toEl DOM nodes are passed to the
function just before the DOM patch operations occurs in LiveView. This allows external
libraries to (re)initialize DOM elements or copy attributes as necessary as LiveView
performs its own patch operations. The update operation cannot be cancelled or deferred,
and the return value is ignored. For example, the following option could be used to add
Alpine.js support to your project:
let liveSocket = new LiveSocket("/live", Socket, {
  ...,
  dom: {
    onBeforeElUpdated(from, to){
      if(from.__x){ window.Alpine.clone(from.__x, to) }
    }
  },
})

Components are a mechanism to compartmentalize state, markup, and
events in LiveView.
Components are defined by using Phoenix.LiveComponent and are used
by calling Phoenix.LiveView.Helpers.live_component/3 in a parent LiveView.
Components run inside the LiveView process, but may have their own
state and event handling.
The simplest component only needs to define a render function:
defmodule HeroComponent do
  # If you generated an app with mix phx.new --live,
  # the line below would be: use MyAppWeb, :live_component
  use Phoenix.LiveComponent

  def render(assigns) do
    ~L"""
    <div class="hero"><%= @content %></div>
    """
  end
end
When use Phoenix.LiveComponent is used, all functions in
Phoenix.LiveView are imported. A component can be invoked as:
<%= live_component @socket, HeroComponent, content: @content %>
Components come in two shapes, stateless or stateful. The component
above is a stateless component. Of course, the component above is not
any different compared to a regular function. However, as we will see,
components do provide their own exclusive feature set.
Stateless components life-cycle
When live_component is called, the following callbacks will be invoked
in the component:
mount(socket) -> update(assigns, socket) -> render(assigns)
First mount/1 is called only with the socket. mount/1 can be used
to set any initial state. Then update/2 is invoked with all of the
assigns given to live_component/3. The default implementation of
update/2 simply merges all assigns into the socket. Then, after the
component is updated, render/1 is called with all assigns.
A stateless component is always mounted, updated, and rendered whenever
the parent template changes. That's why they are stateless: no state
is kept after the component.
However, any component can be made stateful by passing an :id assign.
Stateful components life-cycle
A stateful component is a component that receives an :id on live_component/3:
<%= live_component @socket, HeroComponent, id: :hero, content: @content %>
Stateful components are identified by the component module and their ID.
Therefore, two different component modules with the same ID are different
components. This means we can often tie the component ID to some application
based ID:
<%= live_component @socket, UserComponent, id: @user.id, user: @user %>
Also note the given :id is not necessarily used as the DOM ID. If you
want to set a DOM ID, it is your responsibility to set it when rendering:
defmodule UserComponent do
  use Phoenix.LiveComponent

  def render(assigns) do
    ~L"""
    <div id="user-<%= @id %>" class="user"><%= @user.name %></div>
    """
  end
end
In stateful components, mount/1 is called only once, when the
component is first rendered. For each rendering, the optional
preload/1 and update/2 callbacks are called before render/1.
So on first render, the following callbacks will be invoked:
preload(list_of_assigns) -> mount(socket) -> update(assigns, socket) -> render(assigns)
On subsequent renders, these callbacks will be invoked:
preload(list_of_assigns) -> update(assigns, socket) -> render(assigns)
Targeting Component Events
Stateful components can also implement the handle_event/3 callback
that works exactly the same as in LiveView. For a client event to
reach a component, the tag must be annotated with a phx-target.
If you want to send the event to yourself, you can simply use the
@myself assign, which is an internal unique reference to the
component instance:
<a href="#" phx-click="say_hello" phx-target="<%= @myself %>">
  Say hello!
</a>
Note @myself is not set for stateless components, as they cannot
receive events.
If you want to target another component, you can also pass an ID
or a class selector to any element inside the targeted component.
For example, if there is a UserComponent with the DOM ID of user-13,
using a query selector, we can send an event to it with:
<a href="#" phx-click="say_hello" phx-target="#user-13">
  Say hello!
</a>
In both cases, handle_event/3 will be called with the
"say_hello" event. When handle_event/3 is called for a component,
only the diff of the component is sent to the client, making them
extremely efficient.
Any valid query selector for phx-target is supported, provided that the
matched nodes are children of a LiveView or LiveComponent, for example
to send the close event to multiple components:
<a href="#" phx-click="close" phx-target="#modal, #sidebar">
  Dismiss
</a>
Preloading and update
Every time a stateful component is rendered, both preload/1 and
update/2 are called. To understand why both callbacks are necessary,
imagine that you implement a component and the component needs to load
some state from the database. For example:
<%= live_component @socket, UserComponent, id: user_id %>
A possible implementation would be to load the user on the update/2
callback:
def update(assigns, socket) do
  user = Repo.get! User, assigns.id
  {:ok, assign(socket, :user, user)}
end
However, the issue with said approach is that, if you are rendering
multiple user components in the same page, you have a N+1 query problem.
The preload/1 callback helps address this problem as it is invoked
with a list of assigns for all components of the same type. For example,
instead of implementing update/2 as above, one could implement:
def preload(list_of_assigns) do
  list_of_ids = Enum.map(list_of_assigns, & &1.id)

  users =
    from(u in User, where: u.id in ^list_of_ids, select: {u.id, u})
    |> Repo.all()
    |> Map.new()

  Enum.map(list_of_assigns, fn assigns ->
    Map.put(assigns, :user, users[assigns.id])
  end)
end
Now only a single query to the database will be made. In fact, the
preloading algorithm is a breadth-first tree traversal, which means
that even for nested components, the amount of queries are kept to
a minimum.
Finally, note that preload/1 must return an updated list_of_assigns,
keeping the assigns in the same order as they were given.
Managing state
Now that we have learned how to define and use components, as well as
how to use preload/1 as a data loading optimization, it is important
to talk about how to manage state in components.
Generally speaking, you want to avoid both the parent LiveView and the
LiveComponent working on two different copies of the state. Instead, you
should assume only one of them to be the source of truth. Let's discuss
the two different approaches in detail.
Imagine a scenario where LiveView represents a board with each card in
it as a separate component. Each card has a form that allows to update
its title directly in the component. We will see how to organize the
data flow keeping either the view or the component as the source of truth.
LiveView as the source of truth
If the LiveView is the source of truth, it will be responsible
for fetching all of the cards in a board. Then it will call live_component/3
for each card, passing the card struct as argument to CardComponent:
<%= for card <- @cards do %>
  <%= live_component @socket, CardComponent, card: card, id: card.id, board_id: @id %>
<% end %>
Now, when the user submits a form inside the CardComponent to update the
card, CardComponent.handle_event/3 will be triggered. However, if the
update succeeds, you must not change the card struct inside the component.
If you do so, the card struct in the component will get out of sync with
the LiveView. Since the LiveView is the source of truth, you should instead
tell the LiveView that the card was updated.
Luckily, because the component and the view run in the same process,
sending a message from the component to the parent LiveView is as simple
as sending a message to self():
defmodule CardComponent do
  ...
  def handle_event("update_title", %{"title" => title}, socket) do
    send self(), {:updated_card, %{socket.assigns.card | title: title}}
    {:noreply, socket}
  end
end
The LiveView then receives this event using handle_info:
defmodule BoardView do
  ...
  def handle_info({:updated_card, card}, socket) do
    # update the list of cards in the socket
    {:noreply, updated_socket}
  end
end
As the list of cards in the parent socket was updated, the parent
will be re-rendered, sending the updated card to the component.
So in the end, the component does get the updated card, but always
driven from the parent.
Alternatively, instead of having the component directly send a
message to the parent, the component could broadcast the update
using Phoenix.PubSub. Such as:
defmodule CardComponent do
  ...
  def handle_event("update_title", %{"title" => title}, socket) do
    message = {:updated_card, %{socket.assigns.card | title: title}}
    Phoenix.PubSub.broadcast(MyApp.PubSub, board_topic(socket), message)
    {:noreply, socket}
  end

  defp board_topic(socket) do
    "board:" <> socket.assigns.board_id
  end
end
As long as the parent LiveView subscribes to the "board:ID" topic,
it will receive updates. The advantage of using PubSub is that we get
distributed updates out of the box. Now, if any user connected to the
board changes a card, all other users will see the change.
LiveComponent as the source of truth
If the component is the source of truth, then the LiveView must no
longer fetch all of the cards structs from the database. Instead,
the view must only fetch all of the card ids and render the component
only by passing the IDs:
<%= for card_id <- @card_ids do %>
  <%= live_component @socket, CardComponent, id: card_id, board_id: @id %>
<% end %>
Now, each CardComponent loads their own card. Of course, doing so per
card would be expensive and lead to N queries, where N is the number
of components, so we must use the preload/1 callback to make it
efficient.
Once all card components are started, they can fully manage each
card as a whole, without concerning themselves with the parent LiveView.
However, note that components do not have a handle_info/2 callback.
Therefore, if you want to track distributed changes on a card, you
must have the parent LiveView receive those events and redirect them
to the appropriate card. For example, assuming card updates are sent
to the "board:ID" topic, and that the board LiveView is subscribed to
said topic, one could do:
def handle_info({:updated_card, card}, socket) do
  send_update CardComponent, id: card.id, board_id: socket.assigns.id
  {:noreply, socket}
end
With send_update, the CardComponent given by id will be invoked,
triggering both preload and update callbacks, which will load the
most up to date data from the database.
Live component blocks
When live_component is invoked, it is also possible to pass a do/end
block:
<%= live_component @socket, GridComponent, entries: @entries do %>
  New entry: <%= @entry %>
<% end %>
The do/end will be available as an anonymous function in an assign named
@inner_content. The anonymous function must be invoked passing a new set
of assigns that will be merged into the user assigns. For example, the grid
component above could be implemented as:
defmodule GridComponent do
  use Phoenix.LiveComponent

  def render(assigns) do
    ~L"""
    <div class="grid">
      <%= for entry <- @entries do %>
        <div class="column">
          <%= @inner_content.(entry: entry) %>
        </div>
      <% end %>
    </div>
    """
  end
end
Where the :entry assign was injected into the do/end block.
Note the @inner_content assign is also passed to update/2
along all other assigns. So if you have a custom update/2
implementation, make sure to assign it to the socket like so:
def update(%{inner_content: inner_content}, socket) do
  {:ok, assign(socket, inner_content: inner_content)}
end
The above approach is the preferred one when passing blocks to do/end.
However, if you are outside of a .leex template and you want to invoke a
component passing do/end blocks, you will have to explicitly handle the
assigns by giving it a clause:
live_component @socket, GridComponent, entries: @entries do
  new_assigns -> "New entry: " <> new_assigns[:entry]
end
Live patches and live redirects
A template rendered inside a component can use live_patch and
live_redirect calls. The live_patch is always handled by the parent
LiveView, as components do not provide handle_params.
Limitations
Components require at least one HTML tag
Components must only contain HTML tags at their root. At least one HTML
tag must be present. It is not possible to have components that render
only text or text mixed with tags at the root.
Change tracking requirement
Another limitation of components is that they must always be change
tracked. For example, if you render a component inside form_for, like
this:
<%= form_for @changeset, "#", fn f -> %>
  <%= live_component @socket, SomeComponent, f: f %>
<% end %>
The component ends up enclosed by the form markup, where LiveView
cannot track it. In such cases, you may receive an error such as:
** (ArgumentError) cannot convert component SomeComponent to HTML.
A component must always be returned directly as part of a LiveView template
In this particular case, this can be addressed by using the form_for
variant without anonymous functions:
<%= f = form_for @changeset, "#" %>
  <%= live_component @socket, SomeComponent, f: f %>
</form>
This issue can also happen with other helpers, such as content_tag:
<%= content_tag :div do %>
  <%= live_component @socket, SomeComponent, f: f %>
<% end %>
In this case, the solution is to not use content_tag and rely on LiveEEx
to build the markup.
SVG support
Given components compartmentalize markup on the server, they are also
rendered in isolation on the client, which provides great performance
benefits on the client too.
However, when rendering components on the client, the client needs to
choose the mime type of the component contents, which defaults to HTML.
This is the best default but in some cases it may lead to unexpected
results.
For example, if you are rendering SVG, the SVG will be interpreted as
HTML. This may work just fine for most components but you may run into
corner cases. For example, the <image> SVG tag may be rewritten to
the <img> tag, since <image> is an obsolete HTML tag.
Luckily, there is a solution to this problem. Since SVG allows <svg>
tags to be nested, you can wrap the component content into an <svg>
tag. This will ensure that it is correctly interpreted by the browser.

The struct representing an internal unique reference to the component instance,
available as the @myself assign in stateful components.
Read more about the uses of @myself in the Phoenix.LiveComponent docs.

LiveView provides rich, real-time user experiences with
server-rendered HTML.
The LiveView programming model is declarative: instead of
saying "once event X happens, change Y on the page",
events in LiveView are regular messages which may cause
changes to its state. Once the state changes, LiveView will
re-render the relevant parts of its HTML template and push it
to the browser, which updates itself in the most efficient
manner. This means developers write LiveView templates as
any other server-rendered HTML and LiveView does the hard
work of tracking changes and sending the relevant diffs to
the browser.
At the end of the day, a LiveView is nothing more than a
process that receives events as messages and updates its
state. The state itself is nothing more than functional
and immutable Elixir data structures. The events are either
internal application messages (usually emitted by Phoenix.PubSub)
or sent by the client/browser.
LiveView is first rendered statically as part of regular
HTTP requests, which provides quick times for "First Meaningful
Paint", in addition to helping search and indexing engines.
Then a persistent connection is established between client and
server. This allows LiveView applications to react faster to user
events as there is less work to be done and less data to be sent
compared to stateless requests that have to authenticate, decode, load,
and encode data on every request. The flipside is that LiveView
uses more memory on the server compared to stateless requests.
Use cases
There are many use cases where LiveView is an excellent
fit right now:
	Handling of user interaction and inputs, buttons, and
forms - such as input validation, dynamic forms,
autocomplete, etc;

	Events and updates pushed by server - such as
notifications, dashboards, etc;

	Page and data navigation - such as navigating between
pages, pagination, etc can be built with LiveView
using the excellent live navigation feature set.
This reduces the amount of data sent over the wire,
gives developers full control over the LiveView
life-cycle, while controlling how the browser
tracks those changes in state;


There are also use cases which are a bad fit for LiveView:
	Animations - animations, menus, and general events
that do not need the server in the first place are a
bad fit for LiveView, as they can be achieved purely
with CSS and/or CSS transitions;

Life-cycle
A LiveView begins as a regular HTTP request and HTML response,
and then upgrades to a stateful view on client connect,
guaranteeing a regular HTML page even if JavaScript is disabled.
Any time a stateful view changes or updates its socket assigns, it is
automatically re-rendered and the updates are pushed to the client.
You begin by rendering a LiveView typically from your router.
When LiveView is first rendered, the mount/3 callback is invoked
with the current params, the current session and the LiveView socket.
As in a regular request, params contains public data that can be
modified by the user. The session always contains private data set
by the application itself. The mount/3 callback wires up socket
assigns necessary for rendering the view. After mounting, render/1
is invoked and the HTML is sent as a regular HTML response to the
client.
After rendering the static page, LiveView connects from the client
to the server where stateful views are spawned to push rendered updates
to the browser, and receive client events via phx- bindings. Just like
the first rendering, mount/3 is invoked  with params, session,
and socket state, where mount assigns values for rendering. However
in the connected client case, a LiveView process is spawned on
the server, pushes the result of render/1 to the client and
continues on for the duration of the connection. If at any point
during the stateful life-cycle a crash is encountered, or the client
connection drops, the client gracefully reconnects to the server,
calling mount/3 once again.
Example
Before writing your first example, make sure that Phoenix LiveView
is properly installed. If you are just getting started, this can
be easily done by running mix phx.new my_app --live. The phx.new
command with the --live flag will create a new project with
LiveView installed and configured. Otherwise, please follow the steps
in the installation guide before continuing.
A LiveView is a simple module that requires two callbacks: mount/3
and render/1:
defmodule MyAppWeb.ThermostatLive do
  # If you generated an app with mix phx.new --live,
  # the line below would be: use MyAppWeb, :live_view
  use Phoenix.LiveView

  def render(assigns) do
    ~L"""
    Current temperature: <%= @temperature %>
    """
  end

  def mount(_params, %{"current_user_id" => user_id}, socket) do
    temperature = Thermostat.get_user_reading(user_id)
    {:ok, assign(socket, :temperature, temperature)}
  end
end
The render/1 callback receives the socket.assigns and is responsible
for returning rendered content. You can use Phoenix.LiveView.Helpers.sigil_L/2
to inline LiveView templates.
Next, decide where you want to use your LiveView.
You can serve the LiveView directly from your router (recommended):
defmodule MyAppWeb.Router do
  use Phoenix.Router
  import Phoenix.LiveView.Router

  scope "/", MyAppWeb do
    live "/thermostat", ThermostatLive
  end
end
Note: the above assumes there is plug :put_root_layout call
in your router that configures the LiveView layout. This call is
automatically included by mix phx.new --live and described in
the installation guide. If you don't want to configure a root layout,
you must pass layout: {MyAppWeb.LayoutView, "app.html"} as an
option to the live macro above.
Alternatively, you can live_render from any template:
<h1>Temperature Control</h1>
<%= live_render(@conn, MyAppWeb.ThermostatLive) %>
Or you can live_render your view from any controller:
defmodule MyAppWeb.ThermostatController do
  ...
  import Phoenix.LiveView.Controller

  def show(conn, %{"id" => id}) do
    live_render(conn, MyAppWeb.ThermostatLive)
  end
end
When a LiveView is rendered, all of the data currently stored in the
connection session (see Plug.Conn.get_session/1) will be given to
the LiveView.
It is also possible to pass additional session information to the LiveView
through a session parameter:
# In the router
live "/thermostat", ThermostatLive, session: %{"extra_token" => "foo"}

# In a view
<%= live_render(@conn, MyAppWeb.ThermostatLive, session: %{"extra_token" => "foo"}) %>
Notice the :session uses string keys as a reminder that session data
is serialized and sent to the client. So you should always keep the data
in the session to a minimum. For example, instead of storing a User struct,
you should store the "user_id" and load the User when the LiveView mounts.
Once the LiveView is rendered, a regular HTML response is sent. In your
app.js file, you should find the following:
import {Socket} from "phoenix"
import LiveSocket from "phoenix_live_view"

let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})
liveSocket.connect()
After the client connects, mount/3 will be invoked inside a spawned
LiveView process. At this point, you can use connected?/1 to
conditionally perform stateful work, such as subscribing to pubsub topics,
sending messages, etc. For example, you can periodically update a LiveView
with a timer:
defmodule DemoWeb.ThermostatLive do
  use Phoenix.LiveView
  ...

  def mount(_params, %{"current_user_id" => user_id}, socket) do
    if connected?(socket), do: Process.send_after(self(), :update, 30000)

    case Thermostat.get_user_reading(user_id) do
      {:ok, temperature} ->
        {:ok, assign(socket, temperature: temperature, user_id: user_id)}

      {:error, _reason} ->
        {:ok, redirect(socket, to: "/error")}
    end
  end

  def handle_info(:update, socket) do
    Process.send_after(self(), :update, 30000)
    {:ok, temperature} = Thermostat.get_reading(socket.assigns.user_id)
    {:noreply, assign(socket, :temperature, temperature)}
  end
end
We used connected?(socket) on mount to send our view a message every 30s if
the socket is in a connected state. We receive the :update message in the
handle_info/2 callback, just like in an Elixir GenServer, and update our
socket assigns. Whenever a socket's assigns change, render/1 is automatically
invoked, and the updates are sent to the client.
Colocating templates
In the examples above, we have placed the template directly inside the
LiveView:
defmodule MyAppWeb.ThermostatLive do
  use Phoenix.LiveView

  def render(assigns) do
    ~L"""
    Current temperature: <%= @temperature %>
    """
  end
For larger templates, you can place them in a file in the same directory
and same name as the LiveView. For example, if the file above is placed
at lib/my_app_web/live/thermostat_live.ex, you can also remove the
render/1 definition above and instead put the template code at
lib/my_app_web/live/thermostat_live.html.leex.
Alternatively, you can keep the render/1 callback but delegate to an
existing Phoenix.View module in your application. For example:
defmodule MyAppWeb.ThermostatLive do
  use Phoenix.LiveView

  def render(assigns) do
    Phoenix.View.render(MyAppWeb.PageView, "page.html", assigns)
  end
end
In all cases, each assign in the template will be accessible as @assign.
You can learn more about assigns and LiveEEx templates in their own guide.
Bindings
Phoenix supports DOM element bindings for client-server interaction. For
example, to react to a click on a button, you would render the element:
<button phx-click="inc_temperature">+</button>
Then on the server, all LiveView bindings are handled with the handle_event
callback, for example:
def handle_event("inc_temperature", _value, socket) do
  {:ok, new_temp} = Thermostat.inc_temperature(socket.assigns.id)
  {:noreply, assign(socket, :temperature, new_temp)}
end
	Binding	Attributes
	Params	phx-value-*
	Click Events	phx-capture-click, phx-click
	Focus/Blur Events	phx-window-focus, phx-window-blur, phx-focus, phx-blur
	Key Events	phx-window-keyup, phx-window-keydown, phx-keyup, phx-keydown
	Form Events	phx-auto-recover, phx-trigger-action, phx-disable-with, phx-feedback-for, phx-submit, phx-change
	Rate Limiting	phx-throttle, phx-debounce
	DOM Patching	phx-update
	JS Interop	phx-hook

Compartmentalizing markup and events with render, live_render, and live_component
We can render another template directly from a LiveView template by simply
calling render:
render SomeView, "child_template.html", assigns
Where SomeView is a regular Phoenix.View, typically defined in
lib/my_app_web/views/some_view.ex and "child_template.html" is defined
at lib/my_app_web/templates/some_view/child_template.html.leex. As long
as the template has the .leex extension and all assigns are passed,
LiveView change tracking will also work across templates.
When rendering a child template, any of the phx-* events in the child
template will be sent to the LiveView. In other words, similar to regular
Phoenix templates, a regular render call does not start another LiveView.
This means render is useful for sharing markup between views.
If you want to start a separate LiveView from within a LiveView, then you
can call live_render/3 instead of render/3. This child LiveView runs
in a separate process than the parent, with its own mount and handle_event
callbacks. If a child LiveView crashes, it won't affect the parent. If the
parent crashes, all children are terminated.
When rendering a child LiveView, the :id option is required to uniquely
identify the child. A child LiveView will only ever be rendered and mounted
a single time, provided its ID remains unchanged. Updates to a child session
will be merged on the client, but not passed back up until either a crash and
re-mount or a connection drop and recovery. To force a child to re-mount with
new session data, a new ID must be provided.
Given that a LiveView runs on its own process, it is an excellent tool for creating
completely isolated UI elements, but it is a slightly expensive abstraction if
all you want is to compartmentalize markup and events. For example, if you are
showing a table with all users in the system, and you want to compartmentalize
this logic, rendering a separate LiveView for each user, then using a process
per user would likely be too expensive. For these cases, LiveView provides
Phoenix.LiveComponent, which are rendered using live_component/3:
<%= live_component(@socket, UserComponent, id: user.id, user: user) %>
Components have their own mount and handle_event callbacks, as well as their
own state with change tracking support. Components are also lightweight as they
"run" in the same process as the parent LiveView. However, this means an error
in a component would cause the whole view to fail to render. See Phoenix.LiveComponent
for a complete rundown on components.
To sum it up:
	render - compartmentalizes markup
	live_component - compartmentalizes state, markup, and events
	live_render - compartmentalizes state, markup, events, and error isolation

Endpoint configuration
LiveView accepts the following configuration in your endpoint under
the :live_view key:
	:signing_salt (required) - the salt used to sign data sent
to the client

	:hibernate_after (optional) - the idle time in milliseconds allowed in
the LiveView before compressing its own memory and state.
Defaults to 15000ms (15 seconds)


Guides
LiveView has many guides to help you on your journey.
Server-side
These guides focus on server-side functionality:
	Assigns and LiveEEx
	Error and exception handling
	Live Layouts
	Live Navigation
	Security considerations of the LiveView model
	Telemetry
	Using Gettext for internationalization

Client-side
These guides focus on LiveView bindings and client-side integration:
	Bindings
	Form bindings
	DOM patching and temporary assigns
	JavaScript interoperability

    __using__(opts)
  

    Uses LiveView in the current module to mark it a LiveView.





  
    assign(socket, attrs)
  

    See assign/3.





  
    assign(socket, key, value)
  

    Adds key value pairs to socket assigns.





  
    assign_new(socket, key, func)
  

    Assigns a value into the socket only if it does not exist.





  
    clear_flash(socket)
  

    Clears the flash.





  
    clear_flash(socket, key)
  

    Clears a key from the flash.





  
    connected?(socket)
  

    Returns true if the socket is connected.





  
    get_connect_info(socket)
  

    Accesses the connect info from the socket to use on connected mount.





  
    get_connect_params(socket)
  

    Accesses the connect params sent by the client for use on connected mount.





  
    push_event(socket, event, payload)
  

    Pushes an event to the client to be consumed by hooks.





  
    push_patch(socket, opts)
  

    Annotates the socket for navigation within the current LiveView.





  
    push_redirect(socket, opts)
  

    Annotates the socket for navigation to another LiveView.





  
    put_flash(socket, kind, msg)
  

    Adds a flash message to the socket to be displayed on redirect.

    Annotates the socket for redirect to a destination path.

    Asynchronously updates a Phoenix.LiveComponent with new assigns.

      Link to this macro
    
    __using__(opts)

      
       
       View Source
     
      (macro)
  


  

Uses LiveView in the current module to mark it a LiveView.
use Phoenix.LiveView,
  namespace: MyAppWeb,
  container: {:tr, class: "colorized"},
  layout: {MyAppWeb.LayoutView, "live.html"}

  
  Options


	:namespace - configures the namespace the LiveView is in
	:container - configures the container the LiveView will be wrapped in
	:layout - configures the layout the LiveView will be rendered in

  


  
    
      
      Link to this function
    
    assign(socket, attrs)

      
       
       View Source
     
  


  

See assign/3.
  


  
    
      
      Link to this function
    
    assign(socket, key, value)

      
       
       View Source
     
  


  

Adds key value pairs to socket assigns.
A single key value pair may be passed, or a keyword list
of assigns may be provided to be merged into existing
socket assigns.

  
  Examples


iex> assign(socket, :name, "Elixir")
iex> assign(socket, name: "Elixir", logo: "💧")
  


  
    
      
      Link to this function
    
    assign_new(socket, key, func)

      
       
       View Source
     
  


  

Assigns a value into the socket only if it does not exist.
Useful for lazily assigning values and referencing parent assigns.

  
  Referencing parent assigns


When a LiveView is mounted in a disconnected state, the Plug.Conn assigns
will be available for reference via assign_new/3, allowing assigns to
be shared for the initial HTTP request. On connected mount, assign_new/3
will be invoked, and the LiveView will use its session to rebuild the
originally shared assign. Likewise, nested LiveView children have access
to their parent's assigns on mount using assign_new, which allows
assigns to be shared down the nested LiveView tree.

  
  Examples


# controller
conn
|> assign(:current_user, user)
|> LiveView.Controller.live_render(MyLive, session: %{"user_id" => user.id})

# LiveView mount
def mount(_params, %{"user_id" => user_id}, socket) do
  {:ok, assign_new(socket, :current_user, fn -> Accounts.get_user!(user_id) end)}
end
  


  
    
      
      Link to this function
    
    clear_flash(socket)

      
       
       View Source
     
  


  

Clears the flash.

  
  Examples


iex> clear_flash(socket)
  


  
    
      
      Link to this function
    
    clear_flash(socket, key)

      
       
       View Source
     
  


  

Clears a key from the flash.

  
  Examples


iex> clear_flash(socket, :info)
  


  
    
      
      Link to this function
    
    connected?(socket)

      
       
       View Source
     
  


  

Returns true if the socket is connected.
Useful for checking the connectivity status when mounting the view.
For example, on initial page render, the view is mounted statically,
rendered, and the HTML is sent to the client. Once the client
connects to the server, a LiveView is then spawned and mounted
statefully within a process. Use connected?/1 to conditionally
perform stateful work, such as subscribing to pubsub topics,
sending messages, etc.

  
  Examples


defmodule DemoWeb.ClockLive do
  use Phoenix.LiveView
  ...
  def mount(_params, _session, socket) do
    if connected?(socket), do: :timer.send_interval(1000, self(), :tick)

    {:ok, assign(socket, date: :calendar.local_time())}
  end

  def handle_info(:tick, socket) do
    {:noreply, assign(socket, date: :calendar.local_time())}
  end
end
  


  
    
      
      Link to this function
    
    get_connect_info(socket)

      
       
       View Source
     
  


  

Accesses the connect info from the socket to use on connected mount.
Connect info are only sent when the client connects to the server and
only remain available during mount. nil is returned when called in a
disconnected state and a RuntimeError is raised if called after mount.

  
  Examples


First, when invoking the LiveView socket, you need to declare the
connect_info you want to receive. Typically, it includes at least
the session but it may include other keys, such as :peer_data.
See Phoenix.Endpoint.socket/3:
socket "/live", Phoenix.LiveView.Socket,
  websocket: [connect_info: [:peer_data, session: @session_options]]
Those values can now be accessed on the connected mount as
get_connect_info/1:
def mount(_params, _session, socket) do
  if info = get_connect_info(socket) do
    {:ok, assign(socket, ip: info.peer_data.address)}
  else
    {:ok, assign(socket, ip: nil)}
  end
end
  


  
    
      
      Link to this function
    
    get_connect_params(socket)

      
       
       View Source
     
  


  

Accesses the connect params sent by the client for use on connected mount.
Connect params are only sent when the client connects to the server and
only remain available during mount. nil is returned when called in a
disconnected state and a RuntimeError is raised if called after mount.

  
  Reserved params


The following params have special meaning in LiveView:
	"_csrf_token" - the CSRF Token which must be explicitly set by the user
when connecting
	"_mounts" - the number of times the current LiveView is mounted.
It is 0 on first mount, then increases on each reconnect. It resets
when navigating away from the current LiveView or on errors
	"_track_static" - set automatically with a list of all href/src from
tags with the "phx-track-static" annotation in them. If there are no
such tags, nothing is sent


  
  Examples


def mount(_params, _session, socket) do
  {:ok, assign(socket, width: get_connect_params(socket)["width"] || @width)}
end
  


  
    
      
      Link to this function
    
    push_event(socket, event, payload)

      
       
       View Source
     
  


  

Pushes an event to the client to be consumed by hooks.
Note: events will be dispatched to all active hooks on the client who are
handling the given event. Scoped events can be achieved by namespacing
your event names.

  
  Examples


{:noreply, push_event(socket, "scores", %{points: 100, user: "josé"})}}
  


  
    
      
      Link to this function
    
    push_patch(socket, opts)

      
       
       View Source
     
  


  

Annotates the socket for navigation within the current LiveView.
When navigating to the current LiveView, handle_params/3 is
immediately invoked to handle the change of params and URL state.
Then the new state is pushed to the client, without reloading the
whole page while also maintaining the current scroll position.
For live redirects to another LiveView, use push_redirect/2.

  
  Options


	:to - the required path to link to. It must always be a local path
	:replace - the flag to replace the current history or push a new state.
Defaults false.


  
  Examples


{:noreply, push_patch(socket, to: "/")}
{:noreply, push_patch(socket, to: "/", replace: true)}
  


  
    
      
      Link to this function
    
    push_redirect(socket, opts)

      
       
       View Source
     
  


  

Annotates the socket for navigation to another LiveView.
The current LiveView will be shutdown and a new one will be mounted
in its place, without reloading the whole page. This can
also be used to remount the same LiveView, in case you want to start
fresh. If you want to navigate to the same LiveView without remounting
it, use push_patch/2 instead.

  
  Options


	:to - the required path to link to. It must always be a local path
	:replace - the flag to replace the current history or push a new state.
Defaults false.


  
  Examples


{:noreply, push_redirect(socket, to: "/")}
{:noreply, push_redirect(socket, to: "/", replace: true)}
  


  
    
      
      Link to this function
    
    put_flash(socket, kind, msg)

      
       
       View Source
     
  


  

Adds a flash message to the socket to be displayed on redirect.
Note: While you can use put_flash/3 inside a Phoenix.LiveComponent,
components have their own @flash assigns. The @flash assign
in a component is only copied to its parent LiveView if the component
calls push_redirect/2 or push_patch/2.
Note: You must also place the Phoenix.LiveView.Router.fetch_live_flash/2
plug in your browser's pipeline in place of fetch_flash to be supported,
for example:
import Phoenix.LiveView.Router

pipeline :browser do
  ...
  plug :fetch_live_flash
end

  
  Examples


iex> put_flash(socket, :info, "It worked!")
iex> put_flash(socket, :error, "You can't access that page")
  


  
    
      
      Link to this function
    
    redirect(socket, opts)

      
       
       View Source
     
  


  

Annotates the socket for redirect to a destination path.
Note: LiveView redirects rely on instructing client
to perform a window.location update on the provided
redirect location. The whole page will be reloaded and
all state will be discarded.

  
  Options


	:to - the path to redirect to. It must always be a local path
	:external - an external path to redirect to

  


  
    
      
      Link to this function
    
    send_update(module, assigns)

      
       
       View Source
     
  


  

Asynchronously updates a Phoenix.LiveComponent with new assigns.
The component that is updated must be stateful (the :id in the assigns must
match the :id associated with the component) and the component must be
mounted within the current LiveView.
When the component receives the update, the optional
preload/1 callback is invoked, then
the updated values are merged with the component's assigns and
update/2 is called for the updated
component(s).
While a component may always be updated from the parent by updating some
parent assigns which will re-render the child, thus invoking
update/2 on the child component,
send_update/2 is useful for updating a component that entirely manages its
own state, as well as messaging between components mounted in the same
LiveView.
Note: send_update/2 cannot update a LiveComponent that is mounted in a
different LiveView. To update a component in a different LiveView you must
send a message to the LiveView process that the LiveComponent is mounted
within (often via Phoenix.PubSub).

  
  Examples


def handle_event("cancel-order", _, socket) do
  ...
  send_update(Cart, id: "cart", status: "cancelled")
  {:noreply, socket}
end
  


  
    
      
      Link to this function
    
    static_changed?(socket)

      
       
       View Source
     
  


  

Returns true if the socket is connected and the tracked static assets have changed.
This function is useful to detect if the client is running on an outdated
version of the marked static files. It works by comparing the static paths
sent by the client with the one on the server.
Note: this functionality requires Phoenix v1.5.2 or later.
To use this functionality, the first step is to annotate which static files
you want to be tracked by LiveView, with the phx-track-static. For example:
<link phx-track-static rel="stylesheet" href="<%= Routes.static_path(@conn, "/css/app.css") %>"/>
<script defer phx-track-static type="text/javascript" src="<%= Routes.static_path(@conn, "/js/app.js") %>"></script>
Now, whenever LiveView connects to the server, it will send a copy src
or href attributes of all tracked statics and compare those values with
the latest entries computed by mix phx.digest in the server.
The tracked statics on the client will match the ones on the server the
huge majority of times. However, if there is a new deployment, those values
may differ. You can use this function to detect those cases and show a
banner to the user, asking them to reload the page. To do so, first set the
assign on mount:
def mount(params, session, socket) do
  {:ok, assign(socket, static_changed?: static_changed?(socket))}
end
And then in your views:
<%= if @static_change do %>
  <div id="reload-static">
    The app has been updated. Click here to <a href="#" onclick="window.location.reload()">reload</a>.
  </div>
<% end %>
If you prefer, you can also send a JavaScript script that immediately
reloads the page.
Note: only set phx-track-static on your own assets. For example, do
not set it in external JavaScript files:
<script defer phx-track-static type="text/javascript" src="https://ajax.googleapis.com/ajax/libs/jquery/3.4.1/jquery.min.js"></script>
Because you don't actually serve the file above, LiveView will interpret
the static above as missing, and this function will return true.

    transport_pid(socket)

Returns the transport pid of the socket.
Raises ArgumentError if the socket is not connected.

  
  Examples


iex> transport_pid(socket)
#PID<0.107.0>
  


  
    
      
      Link to this function
    
    update(socket, key, func)

      
       
       View Source
     
  


  

Updates an existing key in the socket assigns.
The update function receives the current key's value and
returns the updated value. Raises if the key does not exist.

  
  Examples


iex> update(socket, :count, fn count -> count + 1 end)
iex> update(socket, :count, &(&1 + 1))

      Link to this callback
    
    handle_call(msg, {}, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          handle_call(
  msg :: term(),
  {pid(), reference()},
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}
  | {:reply, term(), Phoenix.LiveView.Socket.t()}

      


  


  
    
      
      Link to this callback
    
    handle_event(event, unsigned_params, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          handle_event(
  event :: binary(),
  unsigned_params(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:noreply, Phoenix.LiveView.Socket.t()}
  | {:reply, map(), Phoenix.LiveView.Socket.t()}

      


  


  
    
      
      Link to this callback
    
    handle_info(msg, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          handle_info(msg :: term(), socket :: Phoenix.LiveView.Socket.t()) ::
  {:noreply, Phoenix.LiveView.Socket.t()}

      


  


  
    
      
      Link to this callback
    
    handle_params(unsigned_params, uri, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          handle_params(
  unsigned_params(),
  uri :: String.t(),
  socket :: Phoenix.LiveView.Socket.t()
) :: {:noreply, Phoenix.LiveView.Socket.t()}

      


  


  
    
      
      Link to this callback
    
    mount(arg1, session, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          mount(
  unsigned_params() | :not_mounted_at_router,
  session :: map(),
  socket :: Phoenix.LiveView.Socket.t()
) ::
  {:ok, Phoenix.LiveView.Socket.t()}
  | {:ok, Phoenix.LiveView.Socket.t(), keyword()}

      


The LiveView entry-point.
For each LiveView in the root of a template, mount/3 is invoked twice:
once to do the initial page load and again to establish the live socket.
It expects three parameters:
	params - a map of string keys which contain public information that
can be set by the user. The map contains the query params as well as any
router path parameter. If the LiveView was not mounted at the router,
this argument is the atom :not_mounted_at_router
	session - the connection session
	socket - the LiveView socket

It must return either {:ok, socket} or {:ok, socket, options}, where
options is one of:
	:temporary_assigns - a keyword list of assigns that are temporary
and must be reset to their value after every render

	:layout - the optional layout to be used by the LiveView


  


  
    
      
      Link to this callback
    
    render(assigns)

      
       
       View Source
     
  


  
      Specs

      
          render(assigns :: Phoenix.LiveView.Socket.assigns()) ::
  Phoenix.LiveView.Rendered.t()

      


  


  
    
      
      Link to this callback
    
    terminate(reason, socket)

      
       
       View Source
     
      (optional)
  


  
      Specs

      
          terminate(reason, socket :: Phoenix.LiveView.Socket.t()) :: term()
when reason: :normal | :shutdown | {:shutdown, :left | :closed | term()}

      


  

        

      
  
    
Phoenix.LiveView.Controller    



      
The Controller for LiveView rendering.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    live_render(conn, view, opts \\ [])
  

    Renders a live view from a Plug request and sends an HTML response.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this function
    
    live_render(conn, view, opts \\ [])

      
       
       View Source
     
  


  

Renders a live view from a Plug request and sends an HTML response.
Before rendering, the @live_module assign will be added to the
connection assigns for reference.

  
  Options


See Phoenix.LiveView.Helpers.live_render/3 for all supported options.

  
  Examples


defmodule ThermostatController do
  ...
  import Phoenix.LiveView.Controller

  def show(conn, %{"id" => thermostat_id}) do
    live_render(conn, ThermostatLive, session: %{
      "thermostat_id" => id,
      "current_user_id" => get_session(conn, :user_id)
    })
  end
end
  

        

      
  
    
Phoenix.LiveView.Helpers    



      
A collection of helpers to be imported into your views.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    live_component(socket, component, assigns \\ [], do_block \\ [])
  

    Renders a Phoenix.LiveComponent within a parent LiveView.





  
    live_flash(other, key)
  

    Returns the flash message from the LiveView flash assign.





  
    live_patch(opts, opts)
  

    Generates a link that will patch the current LiveView.





  
    live_redirect(opts, opts)
  

    Generates a link that will redirect to a new LiveView.





  
    live_render(conn_or_socket, view, opts \\ [])
  

    Renders a LiveView within an originating plug request or
within a parent LiveView.





  
    live_title_tag(title, opts \\ [])
  

    Renders a title tag with automatic prefix/suffix on @page_title updates.





  
    sigil_L(arg, list)
  

    Provides ~L sigil with HTML safe Live EEx syntax inside source files.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
    
  
    
      
      Link to this macro
    
    live_component(socket, component, assigns \\ [], do_block \\ [])

      
       
       View Source
     
      (macro)
  


  

Renders a Phoenix.LiveComponent within a parent LiveView.
While LiveViews can be nested, each LiveView starts its
own process. A LiveComponent provides similar functionality
to LiveView, except they run in the same process as the
LiveView, with its own encapsulated state.
LiveComponent comes in two shapes, stateful and stateless.
See Phoenix.LiveComponent for more information.

  
  Examples


All of the assigns given are forwarded directly to the
live_component:
<%= live_component(@socket, MyApp.WeatherComponent, id: "thermostat", city: "Kraków") %>
Note the :id won't necessarily be used as the DOM ID.
That's up to the component. However, note that the :id has
a special meaning: whenever an :id is given, the component
becomes stateful. Otherwise, :id is always set to nil.
  


  
    
      
      Link to this function
    
    live_flash(other, key)

      
       
       View Source
     
  


  

Returns the flash message from the LiveView flash assign.

  
  Examples


<p class="alert alert-info"><%= live_flash(@flash, :info) %></p>
<p class="alert alert-danger"><%= live_flash(@flash, :error) %></p>
  


  
    
      
      Link to this function
    
    live_patch(opts, opts)

      
       
       View Source
     
  


  

Generates a link that will patch the current LiveView.
When navigating to the current LiveView, c:handle_params/3 is
immediately invoked to handle the change of params and URL state.
Then the new state is pushed to the client, without reloading the
whole page while also maintaining the current scroll position.
For live redirects to another LiveView, use live_redirect/2.

  
  Options


	:to - the required path to link to.
	:replace - the flag to replace the current history or push a new state.
Defaults false.

All other options are forwarded to the anchor tag.

  
  Examples


<%= live_patch "home", to: Routes.page_path(@socket, :index) %>
<%= live_patch "next", to: Routes.live_path(@socket, MyLive, @page + 1) %>
<%= live_patch to: Routes.live_path(@socket, MyLive, dir: :asc), replace: false do %>
  Sort By Price
<% end %>
  


  
    
      
      Link to this function
    
    live_redirect(opts, opts)

      
       
       View Source
     
  


  

Generates a link that will redirect to a new LiveView.
The current LiveView will be shut down and a new one will be mounted
in its place, without reloading the whole page. This can
also be used to remount the same LiveView, in case you want to start
fresh. If you want to navigate to the same LiveView without remounting
it, use live_patch/2 instead.

  
  Options


	:to - the required path to link to.
	:replace - the flag to replace the current history or push a new state.
Defaults false.

All other options are forwarded to the anchor tag.

  
  Examples


<%= live_redirect "home", to: Routes.page_path(@socket, :index) %>
<%= live_redirect "next", to: Routes.live_path(@socket, MyLive, @page + 1) %>
<%= live_redirect to: Routes.live_path(@socket, MyLive, dir: :asc), replace: false do %>
  Sort By Price
<% end %>
  


    
  
    
      
      Link to this function
    
    live_render(conn_or_socket, view, opts \\ [])

      
       
       View Source
     
  


  

Renders a LiveView within an originating plug request or
within a parent LiveView.

  
  Options


	:session - the map of extra session data to be serialized
and sent to the client. Note that all session data currently in
the connection is automatically available in LiveViews. You
can use this option to provide extra data. Also note that the keys
in the session are strings keys, as a reminder that data has
to be serialized first.
	:container - an optional tuple for the HTML tag and DOM
attributes to be used for the LiveView container. For example:
{:li, style: "color: blue;"}. By default it uses the module
definition container. See the "Containers" section below for more
information.
	:id - both the DOM ID and the ID to uniquely identify a LiveView.
An :id is automatically generated when rendering root LiveViews
but it is a required option when rendering a child LiveView.
	:router - an optional router that enables this LiveView to
perform live navigation. Only a single LiveView in a page may
have the :router set. LiveViews defined at the router with
the live macro automatically have the :router option set.


  
  Examples


# within eex template
<%= live_render(@conn, MyApp.ThermostatLive) %>

# within leex template
<%= live_render(@socket, MyApp.ThermostatLive, id: "thermostat") %>

  
  Containers


When a LiveView is rendered, its contents are wrapped in a container.
By default, said container is a div tag with a handful of LiveView
specific attributes.
The container can be customized in different ways:
	You can change the default container on use Phoenix.LiveView:
use Phoenix.LiveView, container: {:tr, id: "foo-bar"}

	You can override the container tag and pass extra attributes when
calling live_render (as well as on your live call in your router):
live_render socket, MyLiveView, container: {:tr, class: "highlight"}


  


    
  
    
      
      Link to this function
    
    live_title_tag(title, opts \\ [])

      
       
       View Source
     
  


  

Renders a title tag with automatic prefix/suffix on @page_title updates.

  
  Examples


<%= live_title_tag @page_title, prefix: "MyApp – " %>

<%= live_title_tag @page_title, suffix: " – MyApp" %>
  


  
    
      
      Link to this macro
    
    sigil_L(arg, list)

      
       
       View Source
     
      (macro)
  


  

Provides ~L sigil with HTML safe Live EEx syntax inside source files.
iex> ~L"""
...> Hello <%= "world" %>
...> """
{:safe, ["Hello ", "world", "\n"]}
  

        

      
  
    
Phoenix.LiveView.Router    



      
Provides LiveView routing for Phoenix routers.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    fetch_live_flash(conn, _)
  

    Fetches the LiveView and merges with the controller flash.





  
    live(path, live_view, action \\ nil, opts \\ [])
  

    Defines a LiveView route.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    fetch_live_flash(conn, _)

      
       
       View Source
     
  


  

Fetches the LiveView and merges with the controller flash.
Replaces the default :fetch_flash plug used by Phoenix.Router.

  
  Examples


defmodule AppWeb.Router do
  use LiveGenWeb, :router
  import Phoenix.LiveView.Router

  pipeline :browser do
    ...
    plug :fetch_live_flash
  end
  ...
end
  


    
    
  
    
      
      Link to this macro
    
    live(path, live_view, action \\ nil, opts \\ [])

      
       
       View Source
     
      (macro)
  


  

Defines a LiveView route.
A LiveView can be routed to by using the live macro with a path and
the name of the LiveView:
live "/thermostat", ThermostatLive
By default, you can generate a route to this LiveView by using the live_path helper:
live_path(@socket, ThermostatLive)

  
  Actions and live navigation


It is common for a LiveView to have multiple states and multiple URLs.
For example, you can have a single LiveView that lists all articles on
your web app. For each article there is an "Edit" button which, when
pressed, opens up a modal on the same page to edit the article. It is a
best practice to use live navigation in those cases, so when you click
edit, the URL changes to "/articles/1/edit", even though you are still
within the same LiveView. Similarly, you may also want to show a "New"
button, which opens up the modal to create new entries, and you want
this to be reflected in the URL as "/articles/new".
In order to make it easier to recognize the current "action" your
LiveView is on, you can pass the action option when defining LiveViews
too:
live "/articles", ArticleLive.Index, :index
live "/articles/new", ArticleLive.Index, :new
live "/articles/1/edit", ArticleLive.Index, :edit
When an action is given, the generated route helpers are named after
the LiveView itself (in the same way as for a controller). For the example
above, we will have:
article_index_path(@socket, :index)
article_index_path(@socket, :new)
article_index_path(@socket, :edit, 123)
The current action will always be available inside the LiveView as
the @live_action assign, that can be used to render a LiveComponent:
<%= if @live_action == :new do %>
  <%= live_component @socket, MyAppWeb.ArticleLive.FormComponent %>
<% end %>
Or can be used to show or hide parts of the template:
<%= if @live_action == :edit do %>
  <%= render("form.html", user: @user) %>
<% end %>
Note that @live_action will be nil if no action is given on the route definition.

  
  Options


	:session - a map to be merged into the session, for example: %{"my_key" => 123}.
The map keys must be strings.
Can also be a "MFA" (module, function, arguments) tuple. That function will receive
the connection and should return a map (with string keys) to be merged into the session.
For example, {MyModule, :my_function, []} means MyModule.my_function(conn) is called.

	:layout - an optional tuple to specify the rendering layout for the LiveView.
If set, this option will replace the current root layout.

	:container - an optional tuple for the HTML tag and DOM attributes to
be used for the LiveView container. For example: {:li, style: "color: blue;"}.
See Phoenix.LiveView.Helpers.live_render/3 for more information and examples.

	:as - optionally configures the named helper. Defaults to :live when
using a LiveView without actions or defaults to the LiveView name when using
actions.

	:metadata - a map to optional feed metadata used on telemetry events and route info,
for example: %{route_name: :foo, access: :user}.



  
  Examples


defmodule MyApp.Router
  use Phoenix.Router
  import Phoenix.LiveView.Router

  scope "/", MyApp do
    pipe_through [:browser]

    live "/thermostat", ThermostatLive
    live "/clock", ClockLive
    live "/dashboard", DashboardLive, layout: {MyApp.AlternativeView, "app.html"}
  end
end

iex> MyApp.Router.Helpers.live_path(MyApp.Endpoint, MyApp.ThermostatLive)
"/thermostat"
  

        

      
  
    
Phoenix.LiveView.Socket    



      
The LiveView socket for Phoenix Endpoints.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    assigns()
  




  
    fingerprints()
  




  
    t()
  



  

  
    
      Functions
    


  
    connect(params, socket, connect_info)
  

    Connects the Phoenix.Socket for a LiveView client.





  
    id(socket)
  

    Identifies the Phoenix.Socket for a LiveView client.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    assigns()

      
       
       View Source
     
  


  
      Specs

      
          assigns() :: map() | Phoenix.LiveView.Socket.AssignsNotInSocket.t()

      


  


  
    
      
      Link to this type
    
    fingerprints()

      
       
       View Source
     
  


  
      Specs

      
          fingerprints() :: {nil, map()} | {binary(), map()}

      


  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Phoenix.LiveView.Socket{
  assigns: assigns(),
  changed: map(),
  connected?: boolean(),
  endpoint: module(),
  fingerprints: fingerprints(),
  host_uri: URI.t(),
  id: binary(),
  parent_pid: nil | pid(),
  private: map(),
  redirected: nil | tuple(),
  root_pid: pid(),
  root_view: module(),
  router: module(),
  view: module()
}

      


  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    connect(params, socket, connect_info)

      
       
       View Source
     
  


  

Connects the Phoenix.Socket for a LiveView client.
  


  
    
      
      Link to this function
    
    id(socket)

      
       
       View Source
     
  


  

Identifies the Phoenix.Socket for a LiveView client.
  

        

      
  
    
Phoenix.LiveView.Socket.AssignsNotInSocket    



      
Struct for socket.assigns while rendering.
The socket assigns are available directly inside the template
as LiveEEx assigns, such as @foo and @bar. Any assign access
should be done using the assigns in the template where proper change
tracking takes place.

      Types
    


  
    t()
  



  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Phoenix.LiveView.Socket.AssignsNotInSocket{__assigns__: term()}

      


  

        

      
  
    
Phoenix.LiveViewTest    



      
Conveniences for testing Phoenix LiveViews.
In LiveView tests, we interact with views via process
communication in substitution of a browser. Like a browser,
our test process receives messages about the rendered updates
from the view which can be asserted against to test the
life-cycle and behavior of LiveViews and their children.
LiveView Testing
The life-cycle of a LiveView as outlined in the Phoenix.LiveView
docs details how a view starts as a stateless HTML render in a disconnected
socket state. Once the browser receives the HTML, it connects to the
server and a new LiveView process is started, remounted in a connected
socket state, and the view continues statefully. The LiveView test functions
support testing both disconnected and connected mounts separately, for example:
import Plug.Conn
import Phoenix.ConnTest
import Phoenix.LiveViewTest
@endpoint MyEndpoint

test "disconnected and connected mount", %{conn: conn} do
  conn = get(conn, "/my-path")
  assert html_response(conn, 200) =~ "<h1>My Disconnected View</h1>"

  {:ok, view, html} = live(conn)
end

test "redirected mount", %{conn: conn} do
  assert {:error, %{redirect: %{to: "/somewhere"}}} = live(conn, "my-path")
end
Here, we start by using the familiar Phoenix.ConnTest function, get/2 to
test the regular HTTP GET request which invokes mount with a disconnected socket.
Next, live/1 is called with our sent connection to mount the view in a connected
state, which starts our stateful LiveView process.
In general, it's often more convenient to test the mounting of a view
in a single step, provided you don't need the result of the stateless HTTP
render. This is done with a single call to live/2, which performs the
get step for us:
test "connected mount", %{conn: conn} do
  {:ok, _view, html} = live(conn, "/my-path")
  assert html =~ "<h1>My Connected View</h1>"
end
Testing Events
The browser can send a variety of events to a LiveView via phx- bindings,
which are sent to the handle_event/3 callback. To test events sent by the
browser and assert on the rendered side effect of the event, use the
render_* functions:
	render_click/1 - sends a phx-click event and value, returning
the rendered result of the handle_event/3 callback.

	render_focus/2 - sends a phx-focus event and value, returning
the rendered result of the handle_event/3 callback.

	render_blur/1 - sends a phx-blur event and value, returning
the rendered result of the handle_event/3 callback.

	render_submit/1 - sends a form phx-submit event and value, returning
the rendered result of the handle_event/3 callback.

	render_change/1 - sends a form phx-change event and value, returning
the rendered result of the handle_event/3 callback.

	render_keydown/1 - sends a form phx-keydown event and value, returning
the rendered result of the handle_event/3 callback.

	render_keyup/1 - sends a form phx-keyup event and value, returning
the rendered result of the handle_event/3 callback.

	render_hook/3 - sends a hook event and value, returning
the rendered result of the handle_event/3 callback.


For example:
{:ok, view, _html} = live(conn, "/thermo")

assert render_click(view, :inc) =~ "The temperature is: 31℉"

assert render_click(view, :set_temp, 35) =~ "The temperature is: 35℉"

assert render_submit(view, :save, %{deg: 30}) =~ "The temperature is: 30℉"

assert render_change(view, :validate, %{deg: -30}) =~ "invalid temperature"

assert render_keydown(view, :key, :ArrowUp) =~ "The temperature is: 31℉"

assert render_keydown(view, :key, :ArrowDown) =~ "The temperature is: 30℉"
Testing regular messages
LiveViews are GenServer's under the hood, and can send and receive messages
just like any other server. To test the side effects of sending or receiving
messages, simply message the view and use the render function to test the
result:
send(view.pid, {:set_temp, 50})
assert render(view) =~ "The temperature is: 50℉"
Testing components
There are two main mechanisms for testing components. To test stateless
components or just a regular rendering of a component, one can use
render_component/2:
assert render_component(MyComponent, id: 123, user: %User{}) =~
         "some markup in component"
If you want to test how components are mounted by a LiveView and
interact with DOM events, you can use the regular live/2 macro
to build the LiveView with the component and then scope events by
passing the view and a DOM selector in a list:
{:ok, view, html} = live(conn, "/users")
html = view |> element("#user-13 a", "Delete") |> render_click()
refute html =~ "user-13"
refute view |> element("#user-13") |> has_element?()
In the example above, LiveView will lookup for an element with
ID=user-13 and retrieve its phx-target. If phx-target points
to a component, that will be the component used, otherwise it will
fallback to the view.

      Functions
    


  
    assert_patch(view, to, timeout \\ 100)
  

    Asserts a live patch will happen within timeout.





  
    assert_patched(view, to)
  

    Asserts a live patch was performed.





  
    assert_push_event(view, event, payload, timeout \\ 100)
  

    Asserts an event will be pushed within timeout.





  
    assert_redirect(view, to, timeout \\ 100)
  

    Asserts a redirect will happen within timeout.





  
    assert_redirected(view, to)
  

    Asserts a redirect was performed.





  
    assert_reply(view, payload, timeout \\ 100)
  

    Asserts a hook reply was returned from a handle_event callback.





  
    element(view, selector, text_filter \\ nil)
  

    Returns an element to scope a function to.





  
    find_live_child(parent, child_id)
  

    Gets the nested LiveView child by child_id from the parent LiveView.





  
    follow_redirect(reason, conn, to \\ nil)
  

    Follows the redirect from a render_* action.





  
    form(view, selector, form_data \\ %{})
  

    Returns a form element to scope a function to.





  
    has_element?(element)
  

    Checks if the given element exists on the page.





  
    has_element?(view, selector, text_filter \\ nil)
  

    Checks if the given selector with text_filter is on view.





  
    live(conn, path \\ nil)
  

    Spawns a connected LiveView process.





  
    live_children(parent)
  

    Returns the current list of LiveView children for the parent LiveView.





  
    live_isolated(conn, live_view, opts \\ [])
  

    Spawns a connected LiveView process mounted in isolation as the sole rendered element.





  
    page_title(view)
  

    Returns the most recent title that was updated via a page_title assign.





  
    put_connect_info(conn, params)
  

    Puts connect info to be used on LiveView connections.





  
    put_connect_params(conn, params)
  

    Puts connect params to be used on LiveView connections.





  
    render(view)
  

    Returns the HTML string of the rendered view or element.





  
    render_blur(element, value \\ %{})
  

    Sends a blur event given by element and returns the rendered result.





  
    render_blur(view, event, value)
  

    Sends a blur event to the view and returns the rendered result.





  
    render_change(element, value \\ %{})
  

    Sends a form change event given by element and returns the rendered result.





  
    render_change(view, event, value)
  

    Sends a form change event to the view and returns the rendered result.





  
    render_click(element, value \\ %{})
  

    Sends a click event given by element and returns the rendered result.





  
    render_click(view, event, value)
  

    Sends a click event to the view with value and returns the rendered result.





  
    render_component(component, assigns)
  

    Mounts, updates and renders a component.





  
    render_focus(element, value \\ %{})
  

    Sends a focus event given by element and returns the rendered result.





  
    render_focus(view, event, value)
  

    Sends a focus event to the view and returns the rendered result.





  
    render_hook(view_or_element, event, value \\ %{})
  

    Sends a hook event to the view or an element and returns the rendered result.





  
    render_keydown(element, value \\ %{})
  

    Sends a keydown event given by element and returns the rendered result.





  
    render_keydown(view, event, value)
  

    Sends a keydown event to the view and returns the rendered result.





  
    render_keyup(element, value \\ %{})
  

    Sends a keyup event given by element and returns the rendered result.





  
    render_keyup(view, event, value)
  

    Sends a keyup event to the view and returns the rendered result.





  
    render_patch(view, path)
  

    Simulates a live_patch to the given path and returns the rendered result.





  
    render_submit(element, value \\ %{})
  

    Sends a form submit event given by element and returns the rendered result.





  
    render_submit(view, event, value)
  

    Sends a form submit event to the view and returns the rendered result.

      Link to this function
    
    assert_patch(view, to, timeout \\ 100)

      
       
       View Source
     
  


  

Asserts a live patch will happen within timeout.
It always returns :ok. To assert on the flash message,
you can assert on the result of the rendered LiveView.

  
  Examples


render_click(view, :event_that_triggers_patch)
assert_patch view, "/path"
  


  
    
      
      Link to this function
    
    assert_patched(view, to)

      
       
       View Source
     
  


  

Asserts a live patch was performed.
It always returns :ok. To assert on the flash message,
you can assert on the result of the rendered LiveView.

  
  Examples


render_click(view, :event_that_triggers_redirect)
assert_patched view, "/path"
  


    
  
    
      
      Link to this macro
    
    assert_push_event(view, event, payload, timeout \\ 100)

      
       
       View Source
     
      (macro)
  


  

Asserts an event will be pushed within timeout.

  
  Examples


assert_push_event view, "scores", %{points: 100, user: "josé"}
  


    
  
    
      
      Link to this function
    
    assert_redirect(view, to, timeout \\ 100)

      
       
       View Source
     
  


  

Asserts a redirect will happen within timeout.
It returns the flash messages from said redirect, if any.
Note the flash will contain string keys.

  
  Examples


render_click(view, :event_that_triggers_redirect)
flash = assert_redirect view, "/path"
assert flash["info"] == "Welcome"
  


  
    
      
      Link to this function
    
    assert_redirected(view, to)

      
       
       View Source
     
  


  

Asserts a redirect was performed.
It returns the flash messages from said redirect, if any.
Note the flash will contain string keys.

  
  Examples


render_click(view, :event_that_triggers_redirect)
flash = assert_redirected view, "/path"
assert flash["info"] == "Welcome"
  


    
  
    
      
      Link to this macro
    
    assert_reply(view, payload, timeout \\ 100)

      
       
       View Source
     
      (macro)
  


  

Asserts a hook reply was returned from a handle_event callback.

  
  Examples


assert_reply view, "charge", %{amount: 100}, %{result: "ok", transaction_id: _}
  


    
  
    
      
      Link to this function
    
    element(view, selector, text_filter \\ nil)

      
       
       View Source
     
  


  

Returns an element to scope a function to.
It expects the current LiveView, a query selector, and a text filter.
An optional text filter may be given to filter the results by the query
selector. If the text filter is a string or a regex, it will match any
element that contains the string or matches the regex. After the text
filter is applied, only one element must remain, otherwise an error is
raised.
If no text filter is given, then the query selector itself must return
a single element.
assert view
      |> element("#term a:first-child()", "Increment")
      |> render() =~ "Increment</a>"
  


  
    
      
      Link to this function
    
    find_live_child(parent, child_id)

      
       
       View Source
     
  


  

Gets the nested LiveView child by child_id from the parent LiveView.

  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert clock_view = find_live_child(view, "#clock")
assert render_click(clock_view, :snooze) =~ "snoozing"
  


    
  
    
      
      Link to this macro
    
    follow_redirect(reason, conn, to \\ nil)

      
       
       View Source
     
      (macro)
  


  

Follows the redirect from a render_* action.
Imagine you have a LiveView that redirects on a render_click
event. You can make it sure it immediately redirects after the
render_click action by calling follow_redirect/3:
live_view
|> render_click("redirect")
|> follow_redirect(conn)
follow_redirect/3 expects a connection as second argument.
This is the connection that will be used to perform the underlying
request.
If the LiveView redirects with a live redirect, this macro returns
{:ok, live_view, disconnected_html} with the content of the new
LiveView, the same as the live/3 macro. If the LiveView redirects
with a regular redirect, this macro returns {:ok, conn} with the
rendered redirected page. In any other case, this macro raises.
Finally, note that you can optionally assert on the path you are
being redirected to by passing a third argument:
live_view
|> render_click("redirect")
|> follow_redirect(conn, "/redirected/page")
  


    
  
    
      
      Link to this function
    
    form(view, selector, form_data \\ %{})

      
       
       View Source
     
  


  

Returns a form element to scope a function to.
It expects the current LiveView, a query selector, and the form data.
The query selector must return a single element.
The form data will be validated directly against the form markup and
make sure the data you are changing/submitting actually exists, failing
otherwise.

  
  Examples


assert view
      |> form("#term", user: %{name: "hello"})
      |> render_submit() =~ "Name updated"
This function is meant to mimic what the user can actually do, so you cannot
 set hidden input values. However, hidden values can be given when calling
 render_submit/2 or render_change/2, see their docs for examples.
  


  
    
      
      Link to this function
    
    has_element?(element)

      
       
       View Source
     
  


  

Checks if the given element exists on the page.

  
  Examples


assert view |> element("#some-element") |> has_element?()
  


    
  
    
      
      Link to this function
    
    has_element?(view, selector, text_filter \\ nil)

      
       
       View Source
     
  


  

Checks if the given selector with text_filter is on view.
See element/3 for more information.

  
  Examples


assert has_element?(view, "#some-element")
  


    
  
    
      
      Link to this macro
    
    live(conn, path \\ nil)

      
       
       View Source
     
      (macro)
  


  

Spawns a connected LiveView process.
If a path is given, then a regular get(conn, path)
is done and the page is upgraded to a LiveView. If
no path is given, it assumes a previously rendered
%Plug.Conn{} is given, which will be converted to
a LiveView immediately.

  
  Examples


{:ok, view, html} = live(conn, "/path")
assert view.module = MyLive
assert html =~ "the count is 3"

assert {:error, {:redirect, %{to: "/somewhere"}}} = live(conn, "/path")
  


  
    
      
      Link to this function
    
    live_children(parent)

      
       
       View Source
     
  


  

Returns the current list of LiveView children for the parent LiveView.
Children are returned in the order they appear in the rendered HTML.

  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert [clock_view] = live_children(view)
assert render_click(clock_view, :snooze) =~ "snoozing"
  


    
  
    
      
      Link to this macro
    
    live_isolated(conn, live_view, opts \\ [])

      
       
       View Source
     
      (macro)
  


  

Spawns a connected LiveView process mounted in isolation as the sole rendered element.
Useful for testing LiveViews that are not directly routable, such as those
built as small components to be re-used in multiple parents. Testing routable
LiveViews is still recommended whenever possible since features such as
live navigation require routable LiveViews.

  
  Options


	:connect_params - the map of params available in connected mount.
See Phoenix.LiveView.get_connect_params/1 for more information.
	:session - the session to be given to the LiveView

All other options are forwarded to the LiveView for rendering. Refer to
Phoenix.LiveView.Helpers.live_render/3 for a list of supported render
options.

  
  Examples


{:ok, view, html} =
  live_isolated(conn, AppWeb.ClockLive, session: %{"tz" => "EST"})
  


  
    
      
      Link to this function
    
    page_title(view)

      
       
       View Source
     
  


  

Returns the most recent title that was updated via a page_title assign.

  
  Examples


render_click(view, :event_that_triggers_page_title_update)
assert page_title(view) =~ "my title"
  


  
    
      
      Link to this function
    
    put_connect_info(conn, params)

      
       
       View Source
     
  


  

Puts connect info to be used on LiveView connections.
See Phoenix.LiveView.get_connect_info/1.
  


  
    
      
      Link to this function
    
    put_connect_params(conn, params)

      
       
       View Source
     
  


  

Puts connect params to be used on LiveView connections.
See Phoenix.LiveView.get_connect_params/1.
  


  
    
      
      Link to this function
    
    render(view)

      
       
       View Source
     
  


  

Returns the HTML string of the rendered view or element.
If a view is provided, the entire LiveView is rendered. If an
element is provided, only that element is rendered.

  
  Examples


{:ok, view, _html} = live(conn, "/thermo")
assert render(view) =~ ~s|<button id="alarm">Snooze</div>|

assert view
       |> element("#alarm")
       |> render() == "Snooze"
  


    
  
    
      
      Link to this function
    
    render_blur(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a blur event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-blur attribute in it. The event name
given set on phx-blur is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("#inactive")
       |> render_blur() =~ "Tap to wake"
  


  
    
      
      Link to this function
    
    render_blur(view, event, value)

      
       
       View Source
     
  


  

Sends a blur event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_blur(view, :inactive) =~ "Tap to wake"
  


    
  
    
      
      Link to this function
    
    render_change(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a form change event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-change attribute in it. The event name
given set on phx-change is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values.
If you need to pass any extra values or metadata, such as the "_target"
parameter, you can do so by giving a map under the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_change(%{deg: 123}) =~ "123 exceeds limits"

# Passing metadata
{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_change(%{_target: ["deg"], deg: 123}) =~ "123 exceeds limits"
As with render_submit/2, hidden input field values can be provided like so:
refute view
      |> form("#term", user: %{name: "hello"})
      |> render_change(%{"hidden_value" => "example"}) =~ "can't be blank"
  


  
    
      
      Link to this function
    
    render_change(view, event, value)

      
       
       View Source
     
  


  

Sends a form change event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_change(view, :validate, %{deg: 123}) =~ "123 exceeds limits"
  


    
  
    
      
      Link to this function
    
    render_click(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a click event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-click attribute in it. The event name
given set on phx-click is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
If the element is does not have a phx-click attribute but it is
a link (the <a> tag), the link will be followed accordingly:
	if the link is a live_patch, the current view will be patched
	if the link is a live_redirect, this function will return
{:error, {:live_redirect, %{to: url}}}, which can be followed
with follow_redirect/2
	if the link is a regular link, this function will return
{:error, {:redirect, %{to: url}}}, which can be followed
with follow_redirect/2

It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("buttons", "Increment")
       |> render_click() =~ "The temperature is: 30℉"
  


  
    
      
      Link to this function
    
    render_click(view, event, value)

      
       
       View Source
     
  


  

Sends a click event to the view with value and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temperature is: 30℉"
assert render_click(view, :inc) =~ "The temperature is: 31℉"
  


  
    
      
      Link to this macro
    
    render_component(component, assigns)

      
       
       View Source
     
      (macro)
  


  

Mounts, updates and renders a component.
If the component uses the @myself assigns, then an id must
be given to it is marked as stateful.

  
  Examples


assert render_component(MyComponent, id: 123, user: %User{}) =~
         "some markup in component"
  


    
  
    
      
      Link to this function
    
    render_focus(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a focus event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-focus attribute in it. The event name
given set on phx-focus is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given
with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("#inactive")
       |> render_focus() =~ "Tap to wake"
  


  
    
      
      Link to this function
    
    render_focus(view, event, value)

      
       
       View Source
     
  


  

Sends a focus event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_focus(view, :inactive) =~ "Tap to wake"
  


    
  
    
      
      Link to this function
    
    render_hook(view_or_element, event, value \\ %{})

      
       
       View Source
     
  


  

Sends a hook event to the view or an element and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_hook(view, :refresh, %{deg: 32}) =~ "The temp is: 32℉"
If you are pushing events from a hook to a component, then you must pass
an element, created with element/3, as first argument and it must point
to a single element on the page with a phx-target attribute in it:
{:ok, view, _html} = live(conn, "/thermo")
assert view
       |> element("#thermo-component")
       |> render_hook(:refresh, %{deg: 32}) =~ "The temp is: 32℉"
  


    
  
    
      
      Link to this function
    
    render_keydown(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a keydown event given by element and returns the rendered result.
The element is created with element/3 and must point to a single element
on the page with a phx-keydown or phx-window-keydown attribute in it.
The event name given set on phx-keydown is then sent to the appropriate
LiveView (or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values can be given with
the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert view |> element("#inc") |> render_keydown() =~ "The temp is: 31℉"
  


  
    
      
      Link to this function
    
    render_keydown(view, event, value)

      
       
       View Source
     
  


  

Sends a keydown event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_keydown(view, :inc) =~ "The temp is: 31℉"
  


    
  
    
      
      Link to this function
    
    render_keyup(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a keyup event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-keyup or phx-window-keyup attribute
in it. The event name given set on phx-keyup is then sent to the
appropriate LiveView (or component if phx-target is set accordingly).
All phx-value-* entries in the element are sent as values. Extra values
can be given with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert view |> element("#inc") |> render_keyup() =~ "The temp is: 31℉"
  


  
    
      
      Link to this function
    
    render_keyup(view, event, value)

      
       
       View Source
     
  


  

Sends a keyup event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_keyup(view, :inc) =~ "The temp is: 31℉"
  


  
    
      
      Link to this function
    
    render_patch(view, path)

      
       
       View Source
     
  


  

Simulates a live_patch to the given path and returns the rendered result.
  


    
  
    
      
      Link to this function
    
    render_submit(element, value \\ %{})

      
       
       View Source
     
  


  

Sends a form submit event given by element and returns the rendered result.
The element is created with element/3 and must point to a single
element on the page with a phx-submit attribute in it. The event name
given set on phx-submit is then sent to the appropriate LiveView
(or component if phx-target is set accordingly). All phx-value-*
entries in the element are sent as values. Extra values, including hidden
input fields, can be given with the value argument.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")

assert view
       |> element("form")
       |> render_submit(%{deg: 123}) =~ "123 exceeds limits"
To submit a form along with some with hidden input values:
assert view
      |> form("#term", user: %{name: "hello"})
      |> render_submit(%{"hidden_value" => "example"}) =~ "Name updated"
  


  
    
      
      Link to this function
    
    render_submit(view, event, value)

      
       
       View Source
     
  


  

Sends a form submit event to the view and returns the rendered result.
It returns the contents of the whole LiveView or an {:error, redirect}
tuple.

  
  Examples


{:ok, view, html} = live(conn, "/thermo")
assert html =~ "The temp is: 30℉"
assert render_submit(view, :refresh, %{deg: 32}) =~ "The temp is: 32℉"
  

        

      
  
    
Phoenix.LiveViewTest.Element    



      
The struct returned by Phoenix.LiveViewTest.element/3.
The following public fields represent the element:
	selector - The query selector
	text_filter - The text to further filter the element

See the Phoenix.LiveViewTest documentation for usage.
      


  
    
Phoenix.LiveViewTest.View    



      
The struct for testing LiveViews.
The following public fields represent the LiveView:
	id - The DOM id of the LiveView
	module - The module of the running LiveView
	pid - The Pid of the running LiveView
	endpoint - The endpoint for the LiveView

See the Phoenix.LiveViewTest documentation for usage.
      


  
    
Phoenix.LiveView.Component    



      
The struct returned by components in .leex templates.
This component is never meant to be output directly
into the template. It should always be handled by
the diffing algorithm.

    t()

      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Phoenix.LiveView.Component{
  assigns: map(),
  component: module(),
  id: binary()
}

      


  

        

      
  
    
Phoenix.LiveView.Comprehension    



      
The struct returned by for-comprehensions in .leex templates.
See a description about its fields and use cases
in Phoenix.LiveView.Engine docs.

      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Phoenix.LiveView.Comprehension{
  dynamics: [
    [
      iodata()
      | Phoenix.LiveView.Rendered.t()
      | t()
      | Phoenix.LiveView.Component.t()
    ]
  ],
  fingerprint: integer(),
  static: [String.t()]
}

      


  

        

      
  
    
Phoenix.LiveView.Engine    



      
The .leex (Live EEx) template engine that tracks changes.
In the documentation below, we will explain how it works internally.
For user-facing documentation, see Phoenix.LiveView.
Phoenix.LiveView.Rendered
Whenever you render a .leex template, it returns a
Phoenix.LiveView.Rendered structure. This structure has
three fields: :static, :dynamic and :fingerprint.
The :static field is a list of literal strings. This
allows the Elixir compiler to optimize this list and avoid
allocating its strings on every render.
The :dynamic field contains a list of dynamic content.
Each element in the list is either one of:
	iodata - which is the dynamic content
	nil - the dynamic content did not change, see "Tracking changes" below
	another Phoenix.LiveView.Rendered struct, see "Nesting and fingerprinting" below
	a Phoenix.LiveView.Comprehension struct, see "Comprehensions" below
	a Phoenix.LiveView.Component struct, see "Component" below

When you render a .leex template, you can convert the
rendered structure to iodata by intercalating the static
and dynamic fields, always starting with a static entry
followed by a dynamic entry. The last entry will always
be static too. So the following structure:
%Phoenix.LiveView.Rendered{
  static: ["foo", "bar", "baz"],
  dynamic: ["left", "right"]
}
Results in the following content to be sent over the wire
as iodata:
["foo", "left", "bar", "right", "baz"]
This is also what calling Phoenix.HTML.Safe.to_iodata/1
with a Phoenix.LiveView.Rendered structure returns.
Of course, the benefit of .leex templates is exactly
that you do not need to send both static and dynamic
segments every time. So let's talk about tracking changes.
Tracking changes
By default, a .leex template does not track changes.
Change tracking can be enabled by passing a changed
map when invoking the dynamic parts. The map should
contain the name of any changed field as key and the
boolean true as value. If a field is not listed in
:changed, then it is always considered unchanged.
If a field is unchanged and .leex believes a dynamic
expression no longer needs to be computed, its value
in the dynamic list will be nil. This information
can be leveraged to avoid sending data to the client.
Nesting and fingerprinting
Phoenix.LiveView also tracks changes across .leex
templates. Therefore, if your view has this:
<%= render "form.html", assigns %>
Phoenix will be able to track what is static and dynamic
across templates, as well as what changed. A rendered
nested .leex template will appear in the dynamic
list as another Phoenix.LiveView.Rendered structure,
which must be handled recursively.
However, because the rendering of live templates can
be dynamic in itself, it is important to distinguish
which .leex template was rendered. For example,
imagine this code:
<%= if something?, do: render("one.html", assigns), else: render("other.html", assigns) %>
To solve this, all Phoenix.LiveView.Rendered structs
also contain a fingerprint field that uniquely identifies
it. If the fingerprints are equal, you have the same
template, and therefore it is possible to only transmit
its changes.
Comprehensions
Another optimization done by .leex templates is to
track comprehensions. If your code has this:
<%= for point <- @points do %>
  x: <%= point.x %>
  y: <%= point.y %>
<% end %>
Instead of rendering all points with both static and
dynamic parts, it returns a Phoenix.LiveView.Comprehension
struct with the static parts, that are shared across all
points, and a list of dynamics to be interpolated inside
the static parts. If @points is a list with %{x: 1, y: 2}
and %{x: 3, y: 4}, the above expression would return:
%Phoenix.LiveView.Comprehension{
  static: ["\n  x: ", "\n  y: ", "\n"],
  dynamics: [
    ["1", "2"],
    ["3", "4"]
  ]
}
This allows .leex templates to drastically optimize
the data sent by comprehensions, as the static parts
are emitted only once, regardless of the number of items.
The list of dynamics is always a list of iodatas or components,
as we don't perform change tracking inside the comprehensions
themselves. Similarly, comprehensions do not have fingerprints
because they are only optimized at the root, so conditional
evaluation, as the one seen in rendering, is not possible.
The only possible outcome for a dynamic field that returns a
comprehension is nil.
Components
.leex also supports stateful components. Since they are
stateful, they are always handled lazily by the diff algorithm.

    nested_changed_assign?(assigns, changed, all)

Phoenix.LiveView.Rendered    



      
The struct returned by .leex templates.
See a description about its fields and use cases
in Phoenix.LiveView.Engine docs.

      Link to this type
    
    t()

      Specs

      
          t() :: %Phoenix.LiveView.Rendered{
  dynamic:
    (map() | nil ->
       [
         nil
         | iodata()
         | t()
         | Phoenix.LiveView.Comprehension.t()
         | Phoenix.LiveView.Component.t()
       ]),
  fingerprint: integer(),
  static: [String.t()]
}

